

BÀI GIẢNG
HỆ THỐNG THÔNG TIN KẾ TOÁN 3
(EXCEL KẾ TOÁN)

Giảng viên giảng dạy

ThS. Nguyễn Thành Cường
Ngô Xuân Ban
Bùi Mạnh Cường
Đỗ Thị Ly
Phạm Đình Tuấn

Lưu hành nội bộ

MỤC LỤC

Chương 1: Một số kiến thức cơ bản thường dùng trong Excel.....	1
1.1. Cài đặt các thông số thể hiện số trong Excel.....	1
1.2. Một số thao tác về bảng tính.....	1
1.3. Một số hàm thông dụng cơ bản trong Excel.....	3
1.4. Một số chương trình thường trú trong Excel.....	8
Chương 2: Thực hành lập sổ kế toán và báo cáo tài chính	11
2.1. Sơ đồ luồng dữ liệu của Kế toán theo hình thức Nhật ký chung.....	11
2.2. Tổ chức dữ liệu kế toán.....	11

CHƯƠNG 1

MỘT SỐ KIẾN THỨC CƠ BẢN THƯỜNG DÙNG TRONG EXCEL

1.1. CÀI ĐẶT CÁC THÔNG SỐ THỂ HIỆN SỐ TRONG EXCEL

▪ **Định dạng cho tiền tệ với qui ước sử dụng:**

Trong **Control Panel** chọn tùy chọn **Regional settings**, chọn **Currency**, định lại các thông số như sau:

- **Negative number format** : “-” (dấu “-” biểu hiện dấu âm)
- **Currency Symbol** : “Đồng” (tiếp vị ngữ “Đồng” ký hiệu tiền tệ)
- **Decimal Symbol**: “,” (dấu phân cách thập phân là dấu “,”)
- **Digit grouping Symbol**: “.” (dấu phân cách hàng nghìn là dấu “.”)
- **Number of digit in group**: 3 (số lượng phân cách nhóm hàng nghìn là 3 số)

▪ **Định dạng số (Number) với quy ước sử dụng:**

Trong **Control Panel** chọn tùy chọn **Regional settings**. Chọn phiếu **Number**, định lại các thông số như sau:

- **Negative number format**: “-”(dấu “-” biểu diễn dấu âm)
- **Decimal Symol**: “,”(dấu phân cách hàng thập phân là dấu “,”)
- **Digit group Symbol**: “.”(dấu phân cách hàng nghìn là dấu “.”)
- **No. of digits after group** : 2(số chữ thập phân là 2 chữ số)
- **No. of digits in group**: 3 (số lượng phân cách hàng nghìn là 3 số)
- **List separator**: “,” (dấu phân cách trong danh sách (hay phân cách các tham số trong công thức dấu “,”)

1.2. MỘT SỐ THAO TÁC VỀ BẢNG TÍNH:

1.2.1. Chèn thêm Column vào bảng tính

- ☞ Đặt con trỏ vào cột muốn chèn
- ☞ Ra lệnh **Insert** → **Column**

1.2.2. Chèn thêm Row vào bảng tính:

- ☞ Đặt con trỏ vào dòng muốn chèn
- ☞ Ra lệnh **Insert** → **Row**

1.2.3. Chèn thêm sheet (bảng tính) vào workbook:

- ☞ Ra lệnh **Insert/ Work sheet**

1.2.4. Đặt tên cho Worksheet:

- ☞ Chọn **sheet** muốn đặt tên là **sheet** hiện hành
- ☞ **Format** → **Rename** (hoặc double click tại phần tên sheet) – gõ tên **worksheet**.

1.2.5. Dấu (hiện) Row, column, worksheet:

- ☞ Chọn **Row, column** hay **sheet** muốn dấu đi
- ☞ Ra lệnh **format** → (**Row, column** hay **sheet tương ứng**)
- ☞ Muốn dấu thì chọn **Hide** (hiện thì chọn **Unhide**)

1.2.6. Phép tham chiếu ô

Phép tham chiếu dùng để chỉ các ô cụ thể theo sự phối hợp giữa hàng và cột, ví dụ ô A1 là giao điểm giữa cột A và hàng 1.

Nếu muốn chuyển qua lại giữa tham chiếu tương đối, tham chiếu tuyệt đối và tham chiếu hỗn hợp chỉ cần nhấn phím F4.

☞ **Phép tham chiếu ô tuyệt đối**

Ký hiệu đô la (\$) được đặt trước một phối hợp hàng hoặc cột xác định rằng phép tham chiếu mang tính tuyệt đối và sẽ không thay đổi.

Ví dụ: tham chiếu \$A\$1 được dùng trong một công thức sẽ luôn chỉ ô A1, dù cho công thức đặt ở đâu và có sao chép hay không.

☞ **Phép tham chiếu ô tương đối**

Tham chiếu tương đối không có ký hiệu \$. Nó chỉ các ô thông qua sự định vị tương đối.

Ví dụ: Nếu nhập =C8 vào ô D8, công thức này chỉ một ô ở phía trái, trên cùng hàng. Nếu công thức này được sao chép từ ô D8 đến ô D9 thì nó sẽ chỉ đến ô C9 – một ô ở phía trái, trên cùng hàng.


☞ **Phép tham chiếu ô hỗn hợp**

Tham chiếu ô hỗn hợp chứa cả tham chiếu tương đối lẫn tham chiếu tuyệt đối. Nếu phần cột của tham chiếu là tuyệt đối thì cột sẽ không thay đổi, nhưng phần hàng là tương đối sẽ thay đổi. Ngược lại, có thể tham chiếu hỗn hợp với một cột thay đổi và hàng cố định.

Ví dụ: \$G4 là tham chiếu hỗn hợp có cột cố định và hàng thay đổi; G\$4 là tham chiếu hỗn hợp có cột thay đổi và hàng cố định.

1.2.7. Đặt tên khối (Cell Range):

- ☞ Chọn **khối** cần đặt tên bằng cách rê chuột từ ô đầu tiên đến ô cuối cùng của khối. (Cell range). Ra lệnh **Insert → Name → Define**



- ☞ Nhập tên khối vào hộp Name In Work book: (lưu ý: tên khối không có khoảng trắng - chỉ có một từ)

- ☞ Chọn ADD để khai báo

1.2.8. Áp fonts Timesnewroman cho toàn bảng tính:

- ☞ Ra lệnh **Tools → Options... → General**
- ☞ Chọn font **Timesnewroman**, size 12 trong khung **Standard font**:
- ☞ Thoát Excel, sau đó khởi động lại.

1.2.9. Ý nghĩa của thông báo lỗi

Khi Excel không thể ước định chính xác công thức, nó trả về một thông báo lỗi. Tất cả lỗi trong Excel bắt đầu bằng dấu (#). Một số báo lỗi (tùy theo dạng lỗi) hiển thị dưới dạng:

Tên lỗi	Nguyên nhân
#DIV/0	Đây là lỗi divide – by – zero (chia cho 0), thông báo mẫu số bằng 0 do đó không thể thực hiện được phép tính. Exel mặc định các ô trống có giá trị bằng 0, cho nên lỗi này có thể là thực hiện phép chia cho một ô trống
#N/A	Lỗi này có ý nghĩa tùy theo công thức. Có lẽ công thức đã tham chiếu đến một giá trị không tồn tại, ví dụ, trong hàm Vlookup (tìm kiếm), #N/A cũng có thể được sử dụng như một ký tự giữ chỗ (palaceholder) khi dữ liệu chưa có sẵn. Đặc tính lập biểu đồ của Exel bỏ qua #N/A, vì vậy ký hiệu #N/A rất hữu dụng khi bạn vẽ biểu đồ. Nếu bạn để ô trống thì Exel sẽ cho rằng các ô trống có giá trị là 0, và phản ánh sai lệch vào biểu đồ
#NAME?	Lỗi này xảy ra khi Exel không thể nhận diện được tên sử dụng trong công thức. Tên này không tồn tại, hoặc đã bị xoá sơ xuất, hoặc bị đánh vần sai. Cũng có thể bạn quên đặt chuỗi văn bản trong dấu ngoặc kép.
#NUM!	Nghĩa là có vấn đề với một số - số này không thể phiên dịch được vì quá lớn hay quá bé, hoặc nó không tồn tại. Có lẽ bạn đã sử dụng một đối số không phù hợp trong hàm.
#REF!	Lỗi này cho biết có vấn đề với tham chiếu ô, và thường thì do việc xoá hàng hoặc cột – đã được sử dụng trong công thức. Lỗi này cũng có thể do phép tham chiếu từ xa đến một chương trình không chạy, chẳng hạn lỗi DDE.
#VALUE!	Lỗi giá trị do một số nguyên nhân, nhưng thường thì nó đi đôi với việc tính toán ở ký tự (thay vì số), hoặc nhập một đối số không phù hợp cho một hàm.
#NULL!	Lỗi Null cho biết không tồn tại phép giao đối với các dãy ô trong công thức (phần giao rỗng).

1.3. MỘT SỐ HÀM THÔNG DỤNG CƠ BẢN TRONG EXCEL

1.3.1. Hàm Left()

- ☞ **Cú pháp:** =Left(Str,n). Trong đó **Str** là một chuỗi văn bản hoặc một địa chỉ ô chứa giá trị chuỗi, còn **n** là số nguyên dương.
- ☞ **Công dụng:** Hàm trả về giá trị là một chuỗi con được cắt từ bên trái **Str** sang **n** ký tự.
- ☞ **Ví dụ:** =Left("Excel",2) → trả về chuỗi con "Ex"

1.3.2. Hàm Right()

- ☞ **Cú pháp:** =Right(Str,n). Trong đó **Str** là một chuỗi văn bản hoặc một địa chỉ ô chứa giá trị chuỗi, còn **n** là số nguyên dương.
- ☞ **Công dụng:** Hàm trả về giá trị là một chuỗi con được cắt từ bên phải **Str** sang **n** ký tự.
- ☞ **Ví dụ:** =Right ("Excel",2) → trả về chuỗi con "el"

1.3.3. Hàm Mid()

- ☞ **Cú pháp:** =Mid(Str, n, m). Trong đó **Str** là một chuỗi văn bản hoặc một địa chỉ ô chứa giá trị chuỗi, còn **n** và **m** là hai số nguyên dương.
- ☞ **Công dụng:** Hàm trả về giá trị là một chuỗi con được cắt ở giữa **Str** bắt đầu từ ký tự **n** sang **m** ký tự.
- ☞ **Ví dụ:** =Mid("Thực hành kế toán trên Excel", 10, 7) → trả về chuỗi "kế toán"

1.3.4. Hàm If()

- ☞ **Cú pháp:** =IF(**TestValue**, **Value If True**, **Value If False**). Trong đó **TestValue** là biểu thức điều kiện, **Value If True** và **Value If False** là các giá trị sẽ trả về của hàm (hoặc địa chỉ chỉ ô chứa giá trị).
- ☞ **Công dụng:** Khi thực hiện hàm sẽ tính toán và xét biểu thức điều kiện **TestValue**, nếu đúng thì hàm trả về giá trị **Value If True**, nếu sai hàm trả về giá trị **Value If False**.
- ☞ **Ví dụ:** =IF(4>6,4,6) → Hàm trả về số 6.

1.3.5. Hàm Or()

- ☞ **Cú Pháp:** =Or(**Btdk1**, **Btdk2**, **Btdk3**,...). Trong đó: **Btdk1**, **Btdk2**,... là các biểu thức điều kiện trả về giá trị kiểu Logical.
- ☞ **Công dụng:** Hàm trả về giá trị kiểu **Logical**, bằng **True** khi một trong các biểu thức điều kiện trong hàm đúng. Bằng **False** khi tất cả các biểu thức điều kiện đều sai.
- ☞ **Ví dụ:** =Or(2>3, 4>3, 5>7) → Hàm trả về giá trị True
=Or(2>3, 4<3, 5>7) → Hàm trả về giá trị False

1.3.6. Hàm And()

- ☞ **Cú pháp:** =And(**Btdk1**, **Btdk2**, **Btdk3**,...) trong đó: **Btdk1**, **Btdk2**,... là các biểu thức điều kiện trả về giá trị kiểu Logical.
- ☞ **Công dụng:** Hàm trả về giá trị kiểu **Logical**, bằng **False** khi chỉ cần một trong các biểu thức điều kiện trong hàm sai. Bằng **True** khi tất cả các biểu thức điều kiện đều đúng.
- ☞ **Ví dụ:** =And(2>3, 4>3, 5>7) → Hàm trả về giá trị False

1.3.7. Hàm Count()

- ☞ **Cú pháp:** =Count(**Gtrị 1**, **gtrị 2**, **gtrị 3**,...). Trong đó các giá trị có thể là các địa chỉ ô chứa giá trị, một biểu thức, địa chỉ nhóm ô chứa giá trị kiểu số...
- ☞ **Công dụng:** Hàm đếm tổng các giá trị kiểu số liệt kê trong hàm, hay đếm tổng các ô chứa giá trị kiểu số trong hàm.
- ☞ **Ví dụ:** =Count(3, 5, 6, 2) → Kết quả: 4(có 4 số)

1.3.8. Hàm Sum()

- ☞ **Cú pháp:** =Sum(**gtrị 1**, **gtrị 2**, **gtrị 3**...), trong đó các giá trị có thể là các địa chỉ ô chứa giá trị, một biểu thức, địa chỉ nhóm ô chứa giá trị kiểu số...
- ☞ **Công dụng:** Hàm tính tổng các giá trị liệt kê trong hàm hay tính tổng giá trị các ô chứa giá trị trong hàm.
- ☞ **Ví dụ:** =Sum(3, 5, 6, 2) → kết quả: 16

1.3.9. Hàm Round()

- ☞ **Cú pháp:** =Round(**n**,**m**), trong đó **n** là một số lẻ, địa chỉ ô chứa số lẻ: **m** là số nguyên hàng thứ **m**.
- ☞ **Công dụng:** Nếu **m** âm thì làm tròn sang phân nguyên, nếu **m** dương thì hàm làm tròn sang phần thập phân.
- ☞ **Ví dụ:** n=1234,5678 → Round(n,2) = 1234,56; Round(n,-2) = 1200

1.3.10. Hàm Max()

- ☞ **Cú pháp:** =Max(**n1**, **n2**, **n3**...), trong đó **n1**, **n2**, **n3**... là các số, các địa chỉ ô chứa số, địa chỉ nhóm ô chứa số.
- ☞ **Công dụng:** Hàm Max() trả về giá trị kiểu số là số lớn nhất trong dãy số **n1**, **n2**, **n3**...
- ☞ **Ví dụ:** =Max(1, 3, 5, 2, 9) → 9

1.3.11. Hàm Min()

- ☞ **Cú pháp:** =Min(n1, n2, n3...), trong đó n1, n2, n3... là các số, các địa chỉ ô chứa số, địa chỉ nhóm ô chứa số.
- ☞ **Công dụng:** Hàm Min() trả về giá trị kiểu số là số nhỏ nhất trong dãy số n1, n2, n3...
- ☞ **Ví dụ:** =Min(1, 3, 5, 2, 9) → 1

1.3.12. Hàm Average()

- ☞ **Cú pháp:** =Average(n1, n2, n3...) trong đó n1, n2, n3... là các số, các địa chỉ ô chứa số, địa chỉ nhóm ô chứa số.
- ☞ **Công dụng:** Hàm Average() trả về giá trị kiểu số là số bình quân cộng của dãy số n1, n2, n3,...
- ☞ **Ví dụ:** =Average (1, 3, 5, 2, 7) → 3,6; =Average(5, 6) → 6

1.3.13. Hàm Vlookup()

- ☞ **Cú pháp:** Vlookup(LookupValue, Table Array, ColumnIndexNumber, RangeLookup)

Trong đó:

- **LookupValue** là một giá trị (hoặc địa chỉ ô chứa giá trị) mà hàm sẽ mang đi dò tìm trong cột đầu tiên của bảng **TableArray**.
- **TableArray:** là bảng chứa giá trị dò tìm và giá trị lấy ra của hàm. Giá trị dò tìm phải ở cột đầu tiên của bảng, giá trị lấy ra của hàm phải ở các cột từ thứ hai trở đi.
- **ColumnIndexNumber:** là chỉ số cột được chỉ định chứa giá trị trả về trong bảng(chỉ số cột phải >=2, vì giá trị trả về ở các cột từ số 2 trở đi trong bảng).
- **RangeLookup:** là **True()**, hoặc **False()**. Nếu là 0 thì việc dò tìm của hàm phải chính xác, nếu là 1 thì việc dò tìm một giá trị trong bảng không cần chính xác chỉ cần gần đúng là được.
- ☞ **Công dụng:** Nếu tìm thấy giá trị **LookupValue** ở cột đầu tiên của bảng thì hàm trả về giá trị của một ô trong bảng **TableArray** ứng với dòng chứa giá trị dò tìm và cột **ColumnIndexNumber**, ngược lại hàm trả về giá trị **#NA#** (không tìm thấy).

Ta có thể biểu diễn lại cú pháp hàm dưới dạng bảng như sau:

=VLOOKUP("111",

Mã TK	Tên TK
111	TM	
112	TG NH	
113	TĐC	
152	HTK	

,2 ,0)

Với hàm trên, kết quả trả về chuỗi "TM" (giá trị của ô giao nhau giữa dòng 2 và cột 2). Vì dòng 2 của bảng chứa giá trị mang đi dò tìm là "111", chỉ số cột xác định trong hàm là 2. Nếu ta thay giá trị dò tìm là "112" thì hàm sẽ trả về giá trị của ô giao nhau giữa dòng 3 và cột 2 (TGNH). Nếu ta thay giá trị dò tìm là "511" thì hàm sẽ trả về giá trị #N/A (không tìm thấy). Vì TK 511 không có trong bảng dò tìm.

1.3.14. Hàm Match()

- ☞ **Cú pháp:** =Match(LookupValue, LookupArray, MatchType)

Trong đó:

- **LookupValue:** là một giá trị (hoặc địa chỉ ô chứa giá trị) mà hàm sẽ mang đi dò tìm trong mảng **Array**.

- **LookupArray**: là mảng một chiều chứa các giá trị dò tìm (dãy các giá trị) của hàm. Giá trị dò tìm **LookupValue** phải cùng kiểu dữ liệu với các giá trị trong **LookupArray**.
 - **Match Type**: Là **True()** hoặc **False()**. Nếu là 0 thì việc dò tìm của hàm phải chính xác, nếu là 1 thì việc tìm một giá trị trong bảng không cần phải chính xác chỉ cần gần đúng là được.
- ☞ **Công dụng**: Hàm trả về giá trị kiểu số là số thứ tự của phân tử trong mảng chứa **Lookup Value**, ngược lại hàm trả về giá trị **#N/A** (không tìm thấy).

Ta có thể biểu diễn lại cú pháp hàm dưới dạng bảng ví dụ như sau:

=Match ("A01",

A01	A01	B01	B02	B03
------------	-----	-----	-----	-----

0,)

Kết quả trả về của hàm là số 1. Nếu thay giá trị dò tìm là "B02" thì hàm trả về giá trị là số

4. Nếu thay giá trị dò tìm là "B06" thì hàm trả về giá trị là **#N/A** (vì không tìm thấy)

1.3.15. Hàm Index()

☞ **Cú pháp**: **Index(Array, RowNumber, ColumnNumber)**, trong đó:

- **Array**: bảng hai chiều chứa các giá trị trả về của hàm.
 - **Rownumber, ColumnNumber**: là chỉ số dòng và chỉ số cột của ô được chỉ định chứa giá trị trả về trong bảng.
- ☞ **Công dụng**: hàm trả về giá trị của 1 ô trong bảng **Array** là ô giao nhau giữa **RowNumber** và **ColumnNumber**.

Ta có thể biểu diễn lại cú pháp hàm dưới dạng ví dụ như sau:

=INDEX(

11	89	04	42
12	42	57	23
30	65	13	14
15	55	78	83
A	B	C	D

,2,3)

Kết quả trả về **57** (giá trị của ô giao nhau giữa dòng 2 và cột 3). Nếu thay chỉ số dòng là 3 và chỉ số cột là 3 thì kết quả trả về là **13**. Nếu thay chỉ số dòng là 3 và chỉ số cột là 4 thì kết quả trả về là **14**. Nếu thay chỉ số dòng là 5 và chỉ số cột là 4 thì kết quả trả về là "D".

Ta thường dùng kết hợp giữa hàm **Index** với hàm **Match** để tham chiếu đến một giá trị trong bảng, cách dùng như sau:

Ví dụ: Cho bảng đơn giá vận chuyển các loại hàng hoá đến các khu vực như sau:

	A	B	C	D	E	F	G
1		MH01	MH02	MH03	MH04	MH05	MH06
2	TP	12	55	97	24	20	41
3	VT	13	14	19	23	14	12
4	BD	56	45	55	58	44	21
5	MT	85	61	67	80	51	23

☞ Dùng hàm **Index** để tham chiếu đơn giá vận tải của mặt hàng có mã số MH04 đi khu vực BD. Ta thấy khu vực BD ở dòng 3, và mã hàng MH04 ở cột 4 và sử dụng hàm **Index** như sau: **=Index (\$B\$2:\$G\$5,3,4)** và kết quả trả về 58.

☞ Nhưng cách dùng như trên sẽ không linh hoạt khi ta có mã hàng và khu vực thay đổi. Để sử dụng hàm linh hoạt hơn ta kết hợp với hàm **Match**. Hàm **Match** sẽ đóng vai trò xác định chỉ số dòng và chỉ số cột cho hàm **Index**. Ví dụ:

=Index (\$B\$2:\$G\$5,Match("BD",\$A\$2:\$A\$5,0),Match("MH04",\$B\$1:\$G\$1,0))

Kết quả của hàm **Match** thứ nhất trả về số 3, hàm **Match** thứ hai trả về số 4. Do vậy hàm **Index** sẽ trả về giá trị của 1 ô trong bảng **\$B\$2:\$G\$5** là giao nhau giữa dòng 3 và cột 4 (**58**).

☞ Ta cũng thường dùng kết hợp giữa hàm **Vlookup** với hàm **Match** để tham chiếu trên một giá trị trong bảng, trong trường hợp này hàm **Match** sẽ giúp xác định được chỉ số cột **columnIndexNumber** chứa giá trị trả về.

Ví dụ: =Vlookup ("BD",\$B\$1:\$G\$5,Match("BD",\$A\$1:\$G\$1,0),0)

☞ Ta cũng thường dùng kết hợp giữa hàm **Hlookup** với hàm **Match** để tham chiếu trên một giá trị trong bảng, trong trường hợp này hàm **Match** sẽ giúp xác định được chỉ số dòng **RowIndexNumber** chứa giá trị trả về.

Ví dụ: =Hlookup ("MH04",\$B\$1:\$G\$5,Match("BD",\$A\$1:\$A\$5,0),0)

☞ Kết quả của hàm **Index**, **Vlookup**, **Hlookup** là như nhau.

1.3.16. Hàm Sumif()

☞ **Cú pháp: =Sumif(Range, criteria, SumRange)**, trong đó:

- **Range**: cột chứa giá trị để so sánh với điều kiện **Criteria** khi tính toán.
- **Criteria** : Điều kiện tính toán, có kiểu dữ liệu trùng với kiểu dữ liệu của cột **Range**.
- **SumRange**: cột chứa giá trị kiểu số, cột **SumRange** có chiều cao bằng với cột **Range** và tương ứng các giá trị.

☞ **Công dụng**: Hàm tính tổng theo điều kiện.

☞ **Nguyên tắc tính toán của hàm**: Khi thi hành sẽ so sánh điều kiện **Criteria** với các giá trị tổng cột **Range**. Nếu bằng thì lấy giá trị tương ứng (cùng dòng) bên cột **SumRange** cộng lại. Ta có thể biểu diễn lại bằng ví dụ sau:

=Sumif(Range,	Criteria,	Sumrange)												
=Sumif(<table border="1" style="border-collapse: collapse; width: 100px; height: 100px;"> <tr><td>MÃ HÀNG</td></tr> <tr><td>A01</td></tr> <tr><td>A02</td></tr> <tr><td>A03</td></tr> <tr><td>A01</td></tr> <tr><td>A02</td></tr> </table>	MÃ HÀNG	A01	A02	A03	A01	A02	, "A01",	<table border="1" style="border-collapse: collapse; width: 100px; height: 100px;"> <tr><td>SỐ LƯỢNG</td></tr> <tr><td>10</td></tr> <tr><td>10</td></tr> <tr><td>100</td></tr> <tr><td>100</td></tr> <tr><td>20</td></tr> </table>	SỐ LƯỢNG	10	10	100	100	20)
MÃ HÀNG																
A01																
A02																
A03																
A01																
A02																
SỐ LƯỢNG																
10																
10																
100																
100																
20																

Kết quả của hàm trả về số lượng 110. Nếu thay điều kiện thành mã hàng A02 thì kết quả sẽ trả về số lượng 30. Nếu thay điều kiện thành mã hàng A03 thì kết quả trả về số lượng 100.

1.3.17. Hàm Dsum():

☞ **Cú pháp: =Dsum(Database, Field/Number, Criteria)**, trong đó:

- **Database**: Địa chỉ (hoặc tên) bảng dữ liệu nguồn bao gồm cả dòng tiêu đề cột của bảng.
- **Field/Number**: số thứ tự cột trong bảng **DataBase**, cột này có dữ liệu kiểu số được dùng làm cột tính tổng cộng các giá trị trong cột khi điều kiện **Criteria** thoả mãn. **Field/Number** có thể dùng để chỉ số thứ tự cột trong bảng hay dùng tiêu đề cột cần tính của bảng (dùng địa chỉ ô chứa tiêu đề cột).
- **Criteria**: Bảng điều kiện dùng làm căn cứ tính tổng.

☞ **Công dụng:** Hàm **Dsum()** sẽ tính tổng các giá trị trong cột **Field/Number** ứng với dụng chứa điều kiện (thỏa điều kiện) trong bảng điều kiện **criteria**.

☞ **Nguyên lý làm việc:** điều kiện của **criteria** sẽ được so sánh với những giá trị của cột có tiêu đề cột trùng với tiêu đề cột của bảng điều kiện. Nếu bằng thì sẽ lấy giá trị trong cột **FieldNumber** cộng lại, ngược lại sẽ không làm gì cả.

☞ Ta có thể biểu diễn lại hàm **Dsum()** bằng ví dụ sau:

=Dsum(, ,5)

Số CT	TK ghi nợ	Tk ghi có	Số lượng	Số tiền
1	111	112	57	3.000
2	112	111	13	12.000
3	1521	111	78	2.500
4	1521	112	1	90.000

TK ghi nợ	Tk ghi có
1521	111

☞ Kết quả của hàm trả về giá trị: 2.500

☞ Nếu thay bảng điều kiện thành:

TK ghi nợ	Tk ghi có
1521	112
1521	111

☞ Hàm sẽ tính tổng các giá trị trong cột số tiền (cột thứ 5 của bảng) ứng với những định khoản có TK ghi nợ là 1521 đối ứng với 111 hoặc có TK ghi nợ là 1521 và đối ứng với 112. Các định khoản khác có TK ghi nợ là 1521 nhưng tài khoản ghi có khác 111, 112 thì không thỏa mãn điều kiện và không được tính. Kết quả của hàm theo điều kiện trên trả về: **92.500**.

1.4. MỘT CHƯƠNG TRÌNH THƯỜNG TRÚ TRONG EXEL

1.4.1. Dò tìm List và Database

Microsoft Excel cung cấp cho người sử dụng 2 chương trình dùng để dò tìm các mẫu tin (**record**) trong bảng dữ liệu (**Database**) đó là **AutoFiter** và **AdvanceFilter**.

☞ **AutoFiter:** Là chương trình dùng để tìm kiếm những mẫu tin bằng cách che những mẫu tin không thỏa điều kiện và chỉ cho hiện ra trong bảng những mẫu tin thỏa điều kiện mô tả ban đầu. Chương trình này chỉ lọc được các mẫu tin và cho hiện ra tại chỗ mà không thể đem các mẫu tin này sang bảng tính mới, muốn thực hiện việc này cần sử dụng thêm công việc copy.

☞ **AdvanceFilter:** là chương trình dùng để tìm những mẫu tin theo điều kiện mô tả ban đầu. Đặc biệt chương trình này vừa có thể lọc được các mẫu tin tại chỗ và vừa có thể đem các mẫu tin tìm được sang bảng tính mới (**Copy to another Location**).

☞ Như vậy khi bạn cần tìm kiếm một mẫu tin nào đó trong cả hai trường hợp thì bạn phải mô tả được các dữ kiện muốn tìm. Các dữ kiện muốn tìm đó gọi là điều kiện dò tìm (**Criteria**).

1.4.2. Mô tả điều kiện dò tìm (Criteria):

1.4.2.1. Điều kiện có 1 dữ kiện:

Ví dụ: Lập bảng điều kiện dò tìm những định khoản có tài khoản ghi nợ là 111

	A
1	TKGHINO
2	"111"

Ví dụ: Lập bảng điều kiện để dò tìm những định khoản có tài khoản ghi có là 111

	A
1	TKGHICO
2	"111"

1.4.2.2. Điều kiện có nhiều dữ kiện quan hệ và (And):

Ví dụ: Lập bảng điều kiện để dò tìm những định khoản có tài khoản ghi nợ là 112, ghi có là 111:

	A	B	C
1	TKGHINO	TKGHICO	
2	"112"	"111"	

Ví dụ: Lập bảng điều kiện để dò tìm những định khoản có tài khoản ghi nợ là 112 ghi có là 111 và có số tiền phát sinh lớn hơn 100.000:

	A	B	C
1	TKGHINO	TKGHICO	SOTIENPS
2	"112"	"111"	>100.000

Lưu ý: Tất cả các dữ kiện quan hệ And với nhau đều được mô tả trên một dòng.

1.4.2.3. Điều kiện có nhiều dữ kiện quan hệ hoặc (Or):

Ví dụ: Lập bảng điều kiện để dò tìm những định khoản có tài khoản ghi nợ là 111, ghi có là 111 hoặc định khoản có tài khoản.

	A	B	C
1	TKGHINO	TKGHICO	
2	*	"111"	
	"111"	*	

1.4.2.4. Điều kiện có nhiều dữ kiện quan hệ hỗn hợp(And, Or):

Ví dụ: Lập bảng điều kiện để dò tìm những định khoản có tài khoản ghi nợ là 112 ứng với số tiền phát sinh >= 500.000 hoặc định khoản có tài khoản ghi có là 111 ứng với số tiền phát sinh >= 100.000.


	A	B	C
1	TKGHINO	TKGHICO	SOTIENPS
2	*	"111"	>=100.000
3	"112"	*	>=500.000

1.4.2.5. Sử dụng chương trình Autofilter:


Để sử dụng chương trình AutoFilter ta thực hiện các bước sau:

☞ **Bước 1:** Chọn toàn bộ bảng dữ liệu (bao gồm cả dòng tiêu đề). Ví dụ: Chọn sổ nhật ký chung.

☞ **Bước 2:** Chọn **Data/ Filter/ Auto Filter** → trên dòng tiêu đề của sổ nhật ký chung xuất hiện các hộp điều kiện (**Pulldown**), trong mỗi hộp **Pulldown** có chứa các dữ liệu đại diện cho cột đó. Ví dụ:



☞ **Bước 3:** Chọn điều kiện lọc trong từng hộp Pulldown để lọc ra những mẫu tin cần thiết. Cách làm: chọn **Pulldown TKGHNO**



Khi chọn xong ta thấy những định khoản không có tài khoản ghi nợ là 111 sẽ bị che đi, và trong danh sách chỉ còn những định khoản có tài khoản ghi nợ là 111 như điều kiện mô tả.


1.4.2.6. Sắp xếp dữ liệu (sort)

Trong một bảng tính (danh sách) đôi lúc dữ liệu được trình bày và sắp xếp theo ngày phát sinh, theo mã số, hay theo một tiêu thức nào đó để làm cho dữ liệu trong báo cáo trở nên dễ nhìn hơn. Ví dụ: Khi in danh sách lớp học ta thường sắp xếp theo vần Alpha của tên như khi in danh sách trúng tuyển ta phải sắp xếp theo điểm thi...

Việc xác định một tiêu thức sắp xếp tùy thuộc vào yêu cầu mà chính bản thân báo cáo phải đáp ứng. Tiêu thức sắp xếp có thể là đơn hay kép. Ví dụ: Khi in một danh sách nhân viên Công ty thì thường in theo trình tự sắp xếp của Mã phòng ban và mã nhân viên.

Để sắp xếp dữ liệu ta thực hiện các bước sau:

- ☞ **Bước 1:** Chọn toàn bộ bảng (bắt đầu từ dòng tiêu đề các cột đến hết bảng)
- ☞ **Bước 2:** Ra lệnh **Data/ sort** → hiện ra hộp thoại như sau:




☞ **Bước 3:** Trả lời hộp thoại: Chọn khoá (tiêu thức) sắp xếp trong hộp **Sort By** (Ví dụ ColumnA - sắp xếp theo Colum A). Nếu muốn sắp xếp theo tiêu thức thứ hai thì chọn tiêu thức tiếp theo trong hộp **Then By...**

Ta cũng có thể chọn các tiêu thức sắp xếp thông qua tên cột mặc định, trường hợp này ta phải đánh dấu chọn vào **No Header Row** để tắt chế độ thể hiện tiêu đề cột chuyển sang chế độ thể hiện tên cột mặc định.

☞ **Bước 4:** Chọn OK để thi hành.

CHƯƠNG 2 : THỰC HÀNH LẬP SỔ KẾ TOÁN VÀ BÁO CÁO TÀI CHÍNH THEO HÌNH THỨC KẾ TOÁN “NHẬT KÝ CHUNG”

2.1. SƠ ĐỒ LƯỜNG DỮ LIỆU CỦA KẾ TOÁN EXCEL THEO HÌNH THỨC NHẬT KÝ CHUNG:



Qua sơ đồ trên ta thấy tất cả các sổ kế toán , báo cáo tài chính đều nhận thông tin từ BDMTK và SKTM, trong đó:

- Bảng BDMTK cung cấp số dư đầu kỳ của các tài khoản, cung cấp số dư cuối kỳ sau khi đã tổng hợp số phát sinh từ sổ kế toán máy (SKTM) và tính số dư cuối kỳ.
- Sổ kế toán máy (SKTM) cung cấp chi tiết phần nhật ký hàng ngày chính là các bút toán định khoản và những thông tin liên quan

2.2. TỔ CHỨC DỮ LIỆU KẾ TOÁN

2.2.1. Hướng dẫn mở tài khoản sử dụng – số dư đầu kỳ

Tất cả các tài khoản doanh nghiệp sử dụng để định khoản kế toán đều phải đăng ký sẵn và phản ánh số dư đầu kỳ của các tài khoản trong mục danh mục tài khoản. Tạo một Sheet tên là **BDMTK** (bảng danh mục tài khoản), sau đó lập cấu trúc bảng như sau:

	A	B	C	D	E	F	G	H	I	J	K	L
1	BẢNG DANH MỤC TÀI KHOẢN											
2	MATSNV	SOHIEUTK	LOAITK	TENTK	SLTDK	SODDK	SOPSN0	SOPSCO	SOLTCK	SODCK	MATSNV- Điều chỉnh	SODCK- Điều chỉnh
3												
4												
5												
6												
7												
8												
...												

• **Cột MATSNV:** (mã tài sản nguồn vốn) dùng để xác định vị trí của từng tài khoản trong Bảng cân đối kế toán. Trong thành phần của mã tài sản nguồn vốn thì các ký tự được cấu tạo như sau:

- 3 ký tự đầu tiên của **MATSNV** chỉ mã số tổng hợp lớn nhất của Bảng cân đối kế toán (Ví dụ: “100” - Tài sản ngắn hạn).
- 3 ký tự giữa chỉ mã số tổng hợp cấp độ nhỏ hơn (Ví dụ: “110” - Tiền và các khoản tương đương tiền).
- 3 ký tự cuối chỉ thứ tự từng khoản mục nhỏ nhất ghi trong Bảng cân đối kế toán (Ví dụ: “111” - Tiền).

Ví dụ: Số dư của tài khoản 111 được ghi trong bảng cân đối kế toán tại khoản mục có mã số 111 - Tiền, mã số 111 thuộc khoản mục có mã số 110 - Tiền và các khoản tương đương tiền, mã số 110 thuộc khoản mục có mã số 100 - Tài sản ngắn hạn, từ những dữ kiện trên ta đặt MATSNV cho TK 111 là: “100 -110 -111”.

Tương tự như trên ta có thể đặt mã cấp cho các tài khoản khác thuộc loại 1, 2, 3, 4. Còn các tài khoản không có mặt trong bảng cân đối kế toán (loại 5, 6, 7, 8, 9) sẽ không cần đặt mã tài sản nguồn vốn (cột MATSNV bỏ trống).

• **Cột SOHIEUTK:** Mỗi tài khoản có một số hiệu riêng, tài khoản được mở chi tiết cho từng đối tượng theo dõi cụ thể của kế toán chi tiết. Tài khoản mở cho đối tượng nào thì sử dụng mã đối tượng đó ghép với số hiệu tài khoản tương ứng. Tài khoản chi tiết: **<số hiệu tài khoản>.<mã đối tượng>**.

Ví dụ: Mã vật liệu chính M1: Tài khoản chi tiết mở để theo dõi vật liệu này là 1521.M1.

- **Cột loại TK:** Các tài khoản có số phát sinh tăng ghi bên nợ (có số dư bên nợ) thì loại tài khoản quy ước là “**N**”, ngược lại những tài khoản phát sinh tăng ghi bên có (có số dư bên có) thì loại tài khoản quy ước là “**C**”.

- **Cột SLTDK** (số lượng tồn đầu kỳ) dùng để theo dõi số lượng tồn của các tài khoản hàng tồn kho cho từng hàng tồn kho.

- **Cột SODDK** (số dư đầu kỳ) dùng để phản ánh số dư đầu kỳ trên từng tài khoản chi tiết đã mở.

Lưu ý: Nếu một tài khoản có loại TK là “**N**” mà số dư đầu kỳ là số dư **có** thì số dư đầu kỳ phải nhập số dư **âm**, một tài khoản có loại tài khoản là “**C**” mà số dư đầu kỳ là **nợ** thì số dư đầu kỳ cũng phải nhập số **âm**.

Ta đặt tên cho một số vùng để thuận tiện cho việc tham chiếu sau này:

- Đặt tên cho vùng dữ liệu từ **B2:L65536** là **BDMTK** (bảng danh mục tài khoản).
- Đặt tên cho vùng dữ liệu từ **B2:B65536** là **SOHIEUTK** (số hiệu tài khoản).
- Đặt tên cho vùng dữ liệu từ **E2:E65536** là **SLTDK** (số lượng tồn kho đầu kỳ).
- Đặt tên cho vùng dữ liệu từ **F2:F65536** là **SODDK** (số dư đầu kỳ).

2.2.2. Định khoản trên Exel

Tạo một sheet mới tên là SKTM (**sổ kế toán máy**). Có các thành phần như sau:

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	SỔ KẾ TOÁN MÁY												
2	SỐ XÊRI	NGÀY GHI SỔ	SỐ HÓA ĐƠN	SỐ PHIẾU THU/CHI	SỐ PHIẾU NHẬP/XUẤT	NGÀY CHỨNG TỪ	DIỄN GIẢI	TKGHINO	TKGHCO	SỐ LƯỢNG PHÁT SINH	SỐ TIỀN PHÁT SINH	Kiểm tra TK ghi Nợ	Kiểm tra TK ghi Có
3												[1]	[2]
4													
5													
6													
7													
8													
9													

Giải thích các cột:

- Cột **SỐ XÊRI**: Phản ánh số xêri của các hóa đơn GTGT (nhập dạng chuỗi).
- Cột **NGÀY GHI SỔ**: Là ngày định khoản kế toán.
- Cột **SỐ HÓA ĐƠN**: Phản ánh số hóa đơn GTGT (nhập dạng chuỗi).
- Cột **SỐ PHIẾU THU/CHI**: Phản ánh số phiếu thu tiền mặt, chi tiền mặt (nhập dạng chuỗi).
- Cột **SỐ PHIẾU NHẬP/XUẤT**: Phản ánh số phiếu nhập kho/xuất kho (nhập dạng chuỗi).
- Cột **NGÀY CHỨNG TỪ**: Phản ánh chứng từ phát sinh.
- Cột **DIỄN GIẢI**: Ghi trích yếu nội dung nghiệp vụ (nhập dạng chuỗi).

- **Cột TKGHINO, TKGHICO:** Nhập tài khoản ghi nợ và tài khoản ghi có của bút toán định khoản (nhập dạng chuỗi).
- **Cột SỐ LƯỢNG PHÁT SINH:** Ghi nhận số lượng phát sinh (nhập dạng số).
- **Cột SỐ TIỀN PHÁT SINH:** Phản ánh số tiền phát sinh của từng tài khoản trong các bút toán định khoản (nhập dạng số).

❖ **Nội dung:** Tất cả các chứng từ đều được định khoản vào Sổ kế toán máy theo các thông tin trong bảng.

❖ **Cách định khoản:** Khác với định khoản trên các sổ kế toán làm theo hình thức thủ công, định khoản trên Sổ kế toán máy có các đặc điểm riêng phải tuân theo:

- *Phải định khoản theo hình thức đơn giản.*
- *Các bút toán phát sinh của cùng một chứng từ thì các thông tin chung như: số Xêri, ngày ghi sổ, số chứng từ, ngày chứng từ, diễn giải, ... sẽ có chung nội dung.*
- *Một tài khoản đã được mở chi tiết thì tài khoản đó không được sử dụng để định khoản.*

❖ **Để thuận tiện cho việc tổng hợp số liệu sau này ta nên đặt tên một số vùng tham chiếu đến bảng tổng hợp chứng từ gốc như sau:**

- Từ ô **A2:M65536** đặt tên là **SKTM**.
- Từ ô **H3:H65536** đặt tên là **TKGHINO**.
- Từ ô **I3:I65536** tên là **TKGHICO**.
- Từ ô **J3:J65536** tên là **SOLUONGPS**.
- Từ ô **K3:K65536** tên là **SOTIENPS**.

Trong quá trình định khoản kế toán, ta có thể dùng các tài khoản sai đối tượng chi tiết và cũng có khi tài khoản đó chưa mở và đăng ký trong bảng danh mục tài khoản (BDMTK). Nếu như vậy khi tổng hợp số phát sinh vào các tài khoản chi tiết trong bảng BDMTK sẽ không thực hiện được vì không có cơ sở để tổng hợp. Để hạn chế những nhầm lẫn ta có thể lập công thức nhằm phát hiện ra những nhầm lẫn đó:

+ **Cột [1] Kiểm tra tài khoản ghi nợ:** lập công thức như sau:

[1]=VLOOKUP(H3;BDMTK;3;0)

Trong công thức trên, H3 là ô chứa tài khoản ghi nợ đầu tiên trong SKTM.

Hàm VLOOKUP sẽ đem tài khoản trong ô H3 vào cột đầu tiên trong BDMTK tìm kiếm nếu tìm thấy hàm sẽ trả về tên tài khoản tương ứng, nếu không tìm thấy thì hàm trả về thông báo **#NA** nghĩa là tài khoản chưa được khai báo trong bảng BDMTK.

+ Cột [2] **Kiểm tra tài khoản ghi Có**: lập công thức như sau:

[2]=VLOOKUP(I3;BDMTK;3;0)

Trong công thức trên, I3 là ô chứa tài khoản ghi có đầu tiên trong SKTM.

TỔNG HỢP SỐ PHÁT SINH CÁC TÀI KHOẢN CHI TIẾT TỪ BẢNG SKTM VÀO BDMTK:

Ở hai phần trên ta đã thực hiện xong những công việc phải làm ban đầu cho kỳ kế toán đầu tiên tại doanh nghiệp bao gồm việc mở các tài khoản có liên quan sau đó đăng ký chúng trong BDMTK và định khoản các chứng từ kinh tế phát sinh trong kỳ vào Sổ kế toán máy. Ở phần này ta phải thực hiện công việc tổng hợp số phát sinh nợ, có từ tất cả các bút toán định khoản trong bảng SKTM vào từng tài khoản chi tiết tương ứng trong bảng BDMTK. Chi tiết từng chỉ tiêu như sau:

Quay lại bảng BDMTK ta thấy các cột còn bỏ trống là [1], [2], [3], [4]. Đây chính là các cột phải sử dụng công thức để tổng hợp.

✓ **[1] Tổng số phát sinh nợ (SOPSNO)** của các bút toán từ SKTM sang cho từng tài khoản, ta sử dụng hàm SUMIF():

[1]=SUMIF(TKGHINO;B3;SOTIENPS)

Trong công thức trên, ô B3 là địa chỉ ô chứa tài khoản đầu tiên trong BDMTK.

✓ **[2] Tổng số phát sinh có (SOPSCO)** của các bút toán từ SKTM sang cho từng tài khoản, ta sử dụng hàm SUMIF():

[2]=SUMIF(TKGHICO;B3;SOTIENPS)

✓ **[3] Số lượng tồn cuối kỳ của các tài khoản**: nếu các tài khoản là 152, hoặc 153, hoặc 155, hoặc 156 thì mới có số lượng tồn và phải tính theo công thức:

Số lượng tồn cuối kỳ = số lượng tồn đầu kỳ + tổng số lượng nhập – tổng số lượng xuất

[3]= IF(OR(LEFT(B3;3)="152"; LEFT(B3;3)="153"; LEFT(B3;3)="155"; LEFT(B3;3)="156");E3 + SUMIF(TKGHINO;B3;SOLUONGPS) – SUMIF(TKGHICO;B3;SOLUONGPS);0)

Trong công thức trên E3 là ô chứa số lượng tồn đầu kỳ, B3 là ô chứa số hiệu tài khoản đầu tiên của BDMTK.

✓ **[4] Tính số dư cuối kỳ của các tài khoản (SODCK)**

Nếu loại TK (ô C3) là "N" thì số dư cuối kỳ tính bằng công thức:

SODDK (ô F3) + SOPSNO (ô G3) – SOPSCO (ô H3);

Ngược lại thì số dư cuối kỳ được tính bằng công thức:

SODDK (ô F3) + SOPSCO (ô H3) - SOPSNO (ô G3).

[4] = IF(C3="N";F3+G3-H3;F3+H3-G3)

2.2.3. LẬP SỔ NHẬT KÝ

Đối với hình thức kế toán nhật ký chung ngoài sổ nhật ký chung, người ta còn dùng thêm các sổ nhật ký chuyên dùng (sổ nhật ký đặc biệt) như: Nhật ký thu tiền, nhật ký chi tiền, nhật ký mua hàng, nhật ký bán hàng. Mỗi sổ có một chức năng theo dõi, phản ánh các nghiệp vụ kinh tế phát sinh riêng như: Nhật ký thu tiền được mở để theo dõi các nghiệp vụ thu tiền, sổ nhật ký chi tiền được sử dụng để theo dõi các nghiệp vụ chi tiền,.. các nghiệp vụ không được phản ánh trong các sổ nhật ký chuyên dùng sẽ được phản ánh ở sổ nhật ký chung, thông tin để lập các sổ nhật ký được lấy từ sổ kế toán máy (SKTM):

- Nếu bút toán có tài khoản ghi nợ là “111, 112”, thì sẽ được chuyển vào sổ nhật ký thu tiền.
- Nếu bút toán có tài khoản ghi nợ là “131” thì sẽ được chuyển vào sổ nhật ký bán hàng.
- Nếu bút toán có tài khoản ghi có “111, 112”, thì sẽ được chuyển vào sổ nhật ký chi tiền.
- Nếu bút toán có tài khoản ghi có “331” thì sẽ được chuyển vào sổ nhật ký mua hàng.
- Các bút toán còn lại sẽ được ghi vào sổ nhật ký chung.

SỔ NHẬT KÝ THU TIỀN MẶT

Những bút toán nào có tài khoản ghi nợ là 111, 112 thì sẽ được chuyển vào sổ nhật ký thu tiền mặt.



Tạo một sheet mới có tên là **NKTTM**, cấu trúc sổ như sau:

	A	B	C	D	E	F	G	H	I	J	K	L	
1	NHẬT KÝ THU TIỀN MẶT												
2	Tháng Năm												
3													
4	Ngày ghi sổ	Chứng từ		Diễn giải	Ghi nợ TK 111	Ghi có các tài khoản						Tài khoản khác	
5		Số	Ngày			'112	'141	'131	'511	'3331	Số tiền	Số hiệu	
6													
7	1	2	3	4	5	6	7	8	9	10	11	12	
8	[1]	[2]	[3]	[4]	[5]	[6]	[7]	[8]	[9]	[10]	[11]	[12]	
9													
10													
11													
12													
n	Tổng cộng				[13]	[14]	[15]	[16]	[17]	[18]	[19]		

Các ô có [?] ta phải thiết lập công thức, cụ thể là:

✓ **[1] Ngày ghi sổ:**

[1] = IF(LEFT(SKTM!\$H3;3)="111",SKTM!B3;"")

✓ **[2] Số chứng từ:**

[2] = IF(LEFT(SKTM!\$H3;3)="111",SKTM!D3;"")

✓ **[3] Ngày chứng từ:**

[3] = IF(LEFT(SKTM!\$H3;3)="111",SKTM!F3;"")

✓ **[4] Diễn giải:**

[4] = IF(LEFT(SKTM!\$H3;3)="111",SKTM!G3;"")

✓ **[5] Ghi nợ tài khoản 111:**

[5] = IF(LEFT(SKTM!\$H3;3)="111",SKTM!K3;0)

✓ **[6] Ghi có tài khoản 112 đối ứng 111:**

[6] = IF(AND(LEFT(SKTM!\$H3;3)="111";LEFT(SKTM!\$I3;3)=F\$5);\$E8;0)

- ✓ [7] Ghi có tài khoản 141 đối ứng 111:

[7] = IF(AND(LEFT(SKTM!\$H3;3)="111";LEFT(SKTM!\$I3;3)=G\$5);\$E8;0)

- ✓ [8] Ghi có tài khoản 131 đối ứng 111:

[8] = IF(AND(LEFT(SKTM!\$H3;3)="111";LEFT(SKTM!\$I3;3)=H\$5);\$E8;0)

- ✓ [9] Ghi có tài khoản 511 đối ứng 111:

[9] = IF(AND(LEFT(SKTM!\$H3;3)="111";LEFT(SKTM!\$I3;3)=I\$5);\$E8;0)

- ✓ [10] Ghi có tài khoản 3331 đối ứng 111:

[10] = IF(AND(LEFT(SKTM!\$H3;3)="111";LEFT(SKTM!\$I3;4)=J\$5);\$E8;0)

- ✓ [11] Số tiền ghi có các tài khoản khác đối ứng với 111:

[11] = IF(SUM(F8:J8)>0;0;E8)

- ✓ [12] Số hiệu tài khoản ghi có khác đối ứng với 111:

[12] = IF(K8=0;"";SKTM!I3)

- ✓ Copy các công thức trên đến trước hàng tổng cộng (n-1).

- ✓ [13] Tổng cộng số tiền phát sinh nợ tài khoản 111:

[13] = SUM(E8:E_{n-1})

Ghi chú: n: dòng thứ n (dòng tổng cộng).

- ✓ [14], [15], [16], [17], [18], [19] làm tương tự như [13].

Cuối cùng ta cài đặt bộ lọc tự động AutoFilter vào cột diễn giải để lọc ra những dòng có số liệu./.

SỔ NHẬT KÝ CHI TIỀN MẶT.

Những bút toán có tài khoản ghi có "111" thì sẽ được chuyển vào sổ nhật ký chi tiền.



Tạo một sheet có tên là **NKCTM**, cấu trúc sổ như sau:

	A	B	C	D	E	F	G	H	I	J	K	
1	NHẬT KÝ CHI TIỀN MẶT											
2	Tháng Năm											
3												
4	Ngày ghi sổ	Chứng từ		Diễn giải	Ghi có TK 111	Ghi nợ các tài khoản					Tài khoản khác	
5		Số	Ngày			'642	'141	'152	'133	'112	Số tiền	Số hiệu
6												
7	1	2	3	4	5	6	7	8	9	10	11	12
8	[1]	[2]	[3]	[4]	[5]	[6]	[7]	[8]	[9]	[10]	[11]	[12]
9												
10												
11												
12												
n	Tổng cộng				[13]	[14]	[15]	[16]	[17]	[18]	[19]	

Các ô có [?] ta phải thiết lập công thức, cụ thể là:

✓ [1] Ngày ghi sổ:

[1] = IF(LEFT(SKTM!\$I3;3)="111",SKTM!B3;"")

✓ [2] Số chứng từ:

[2] = IF(LEFT(SKTM!\$I3;3)="111",SKTM!D3;"")

✓ [3] Ngày chứng từ:

[3] = IF(LEFT(SKTM!\$I3;3)="111",SKTM!F3;"")

✓ [4] Diễn giải:

[4] = IF(LEFT(SKTM!\$I3;3)="111",SKTM!G3;"")

✓ [5] Ghi có tài khoản 111:

[5] = IF(LEFT(SKTM!\$I3;3)="111",SKTM!K3;"")

✓ [6] Ghi nợ tài khoản 642 đối ứng 111:

[6] = IF(AND(LEFT(SKTM!\$I3;3)="111";LEFT(SKTM!\$H3;3)=F\$5);\$E8;0)

- ✓ [7] Ghi nợ tài khoản 141 đối ứng 111:

[7] = IF(AND(LEFT(SKTM!\$I3;3)="111";LEFT(SKTM!\$H3;3)=G\$5);\$E8;0)

- ✓ [8] Ghi nợ tài khoản 152 đối ứng 111:

[8] = IF(AND(LEFT(SKTM!\$I3;3)="111";LEFT(SKTM!\$H3;3)=H\$5);\$E8;0)

- ✓ [9] Ghi nợ tài khoản 133 đối ứng 111:

[9] = IF(AND(LEFT(SKTM!\$I3;3)="111";LEFT(SKTM!\$H3;3)=I\$5);\$E8;0)

- ✓ [10] Ghi nợ tài khoản 112 đối ứng 111:

[10] = IF(AND(LEFT(SKTM!\$I3;3)="111";LEFT(SKTM!\$H3;3)=J\$5);\$E8;0)

- ✓ [11] Số tiền ghi nợ các tài khoản khác đối ứng với 111:

[11] = IF(SUM(F8:J8)>0;0;E8)

- ✓ [12] Số hiệu tài khoản ghi nợ khác đối ứng với 111:

[12] = IF(K8=0;"";SKTM!H3)

- ✓ Copy các công thức trên đến trước hàng tổng cộng (n-1).

- ✓ [13] Tổng cộng số tiền phát sinh có tài khoản 111:

[13] = SUM(E8:E_{n-1})

Ghi chú: n: dòng thứ n (dòng tổng cộng).

- ✓ [14], [15], [16], [17], [18], [19] làm tương tự như [13].

Cuối cùng ta cài đặt bộ lọc tự động AutoFilter vào cột diễn giải để lọc ra những dòng có số liệu.

SỔ NHẬT KÝ MUA HÀNG



Tạo một sheet có tên là **NKMH**, cấu trúc sổ như sau:

	A	B	C	D	E	F	G	H	I	J	K	L
1	NHẬT KÝ MUA HÀNG											
2	Tháng Năm											
3												
4		Chứng từ			Ghi	Ghi nợ các tài khoản						
5	Ngày ghi sổ	Số	Ngày	Diễn giải	Ghi có TK 331	'1521	'1522	'1523	'153	'133	Tài khoản khác	
6											Số tiền	Số hiệu
7	1	2	3	4	5	6	7	8	9	10	11	12
8	[1]	[2]	[3]	[4]	[5]	[6]	[7]	[8]	[9]	[10]	[11]	[12]
9												
10												
11												
12												
n	Tổng cộng				[13]	[14]	[15]	[16]	[17]	[18]	[19]	

Đặc điểm của sổ nhật ký mua hàng đó là sổ phản ánh các nghiệp vụ liên quan đến việc mua chịu. Do đó, nghiệp vụ có tài khoản ghi có là "331" và tài khoản ghi nợ khác "331" đều là những nghiệp vụ mua chịu, còn các nghiệp vụ có tài khoản ghi có là "331" và tài khoản ghi nợ là "331" là các bút toán chuyển đổi công nợ thì không phản ánh vào nhật ký mua hàng.

Các ô có [?] ta phải thiết lập công thức, cụ thể là:

✓ **[1] Ngày ghi sổ:**

**[1] = IF(AND(LEFT(SKTM!\$I3;3)="331";
LEFT(SKTM!\$H3;3)<>"331");SKTM!B3;"")**

✓ **[2] Số chứng từ:**

**[2] = IF(AND(LEFT(SKTM!\$I3;3)="331";
LEFT(SKTM!\$H3;3)<>"331");SKTM!C3;"")**

✓ **[3] Ngày chứng từ:**

**[3] = IF(AND(LEFT(SKTM!\$I3;3)="331";
LEFT(SKTM!\$H3;3)<>"331");SKTM!F3;"")**

- ✓ [4] Diễn giải:

[4] = IF(AND(LEFT(SKTM!\$I3;3)="331";
LEFT(SKTM!\$H3;3)<>"331");SKTM!G3;"")

- ✓ [5] Ghi có tài khoản 331:

[5] = IF(AND(LEFT(SKTM!I3;3)="331";
LEFT(SKTM!H3;3)<>"331");SKTM!K3;0)

- ✓ [6] Ghi nợ tài khoản 1521 đối ứng 331:

[6] = IF(AND(LEFT(SKTM!\$I3;3)="331";LEFT(SKTM!\$H3;4)=F\$5);\$E8;0)

- ✓ [7] Ghi nợ tài khoản 1522 đối ứng 331:

[7] = IF(AND(LEFT(SKTM!\$I3;3)="331";LEFT(SKTM!\$H3;4)=G\$5);\$E8;0)

- ✓ [8] Ghi nợ tài khoản 1523 đối ứng 331:

[8] = IF(AND(LEFT(SKTM!\$I3;3)="331";LEFT(SKTM!\$H3;4)=H\$5);\$E8;0)

- ✓ [9] Ghi nợ tài khoản 153 đối ứng 331:

[9] = IF(AND(LEFT(SKTM!\$I3;3)="331";LEFT(SKTM!\$H3;3)=I\$5);\$E8;0)

- ✓ [10] Ghi nợ tài khoản 133 đối ứng 331:

[10] = IF(AND(LEFT(SKTM!\$I3;3)="331";LEFT(SKTM!\$H3;3)=J\$5);\$E8;0)

- ✓ [11] Số tiền ghi nợ các tài khoản khác đối ứng với 331:

[11] = IF(SUM(F8:J8)>0;0;E8)

- ✓ [12] Số hiệu tài khoản ghi nợ khác đối ứng với 331:

[12] = IF(K8=0;"";SKTM!H3)

- ✓ Copy các công thức trên đến trước hàng tổng cộng (n-1).

- ✓ [13] Tổng cộng số tiền phát sinh có tài khoản 331:


[13] = SUM(E8:E_{n-1})

Ghi chú: n: dòng thứ n (dòng tổng cộng).

- ✓ **[14], [15], [16], [17], [18], [19] làm tương tự như [13].**

Cuối cùng ta cài đặt bộ lọc tự động AutoFilter vào cột diễn giải để lọc ra những dòng có số liệu.

SỔ NHẬT KÝ BÁN HÀNG



Tạo một sheet có tên là **NKBH**, cấu trúc sổ như sau:

	A	B	C	D	E	F	G	H	I	J	K	L
1	NHẬT KÝ BÁN HÀNG											
2	Tháng Năm											
3												
4		Chứng từ			Ghi	Ghi có các tài khoản						
5	Ngày ghi sổ	Số	Ngày	Diễn giải	Ghi nợ TK 131	'5111	'5112	'5113	'3331	'...	Tài khoản khác	
6											Số tiền	Số hiệu
7	1	2	3	4	5	6	7	8	9	10	11	12
8	[1]	[2]	[3]	[4]	[5]	[6]	[7]	[8]	[9]	[10]	[11]	[12]
9												
10												
11												
12												
n	Tổng cộng				[13]	[14]	[15]	[16]	[17]	[18]	[19]	

Những chứng từ có tài khoản ghi nợ là “131” và tài khoản ghi có khác “131” là những nghiệp vụ bán hàng chịu ta phải chuyển số liệu từ SKTM vào sổ nhật ký bán hàng, còn các bút toán có tài khoản ghi nợ là “131” và tài khoản ghi có bằng “131” là những bút toán chuyển đổi từ nợ dài hạn thành nợ ngắn hạn thì không phản ánh vào sổ nhật ký bán hàng.

Các ô có [?] ta phải thiết lập công thức, cụ thể là:

- ✓ **[1] Ngày ghi sổ:**

[1] = IF(AND(LEFT(SKTM!\$H3;3)=""131"; LEFT(SKTM!\$I3;3)<>"131");SKTM!B3;"")

- ✓ **[2] Số chứng từ:**

[2] = IF(AND(LEFT(SKTM!\$H3;3)=""131"; LEFT(SKTM!\$I3;3)<>"131");SKTM!C3;"")

- ✓ [3] Ngày chứng từ:

[3] = IF(AND(LEFT(SKTM!\$H3;3)="131";
LEFT(SKTM!\$I3;3)<>"131");SKTM!F3;"")

- ✓ [4] Diễn giải:

[4] = IF(AND(LEFT(SKTM!\$H3;3)="131";
LEFT(SKTM!\$I3;3)<>"131");SKTM!G3;"")

- ✓ [5] Ghi nợ tài khoản 131:

[5] = IF(AND(LEFT(SKTM!H3;3)="131";
LEFT(SKTM!I3;3)<>"131");SKTM!K3;0)

- ✓ [6] Ghi có tài khoản 5111 đối ứng 131:

[6] = IF(AND(LEFT(SKTM!\$H3;3)="131";LEFT(SKTM!\$I3;4)=F\$5);\$E8;0)
[7] = IF(AND(LEFT(SKTM!\$H3;3)="131";LEFT(SKTM!\$I3;4)=G\$5);\$E8;0)

- ✓ [7] Ghi có tài khoản 5112 đối ứng 131:

[8] = IF(AND(LEFT(SKTM!\$H3;3)="131";LEFT(SKTM!\$I3;4)=H\$5);\$E8;0)

- ✓ [8] Ghi có tài khoản 5113 đối ứng 131:

- ✓ [9] Ghi có tài khoản 3331 đối ứng 131:

- ✓ [10] Ghi có tài khoản ... đối ứng 131:

[9] = IF(AND(LEFT(SKTM!\$H3;3)="131";LEFT(SKTM!\$I3;4)=I\$5);\$E8;0)

[10] = IF(AND(LEFT(SKTM!\$H3;3)="131";LEFT(SKTM!\$I3;4)=J\$5);\$E8;0)

- ✓ [11] Số tiền ghi có các tài khoản khác đối ứng với 131:

[11] = IF(SUM(F8:J8)>0;0;E8)

- ✓ [12] Số hiệu tài khoản ghi có khác đối ứng với 131:

[12] = IF(K8=0;"";SKTM!I3)

- ✓ **Copy các công thức trên đến trước hàng tổng cộng (n-1).**
- ✓ **[13] Tổng cộng số tiền phát sinh nợ tài khoản 131:**

$$[13] = \text{SUM}(E8:E_{n-1})$$

Ghi chú: n: dòng thứ n (dòng tổng cộng).

- ✓ **[14], [15], [16], [17], [18], [19] làm tương tự như [13].**

Cuối cùng ta cài đặt bộ lọc tự động AutoFilter vào cột diễn giải để lọc ra những dòng có số liệu.

SỔ NHẬT KÝ CHUNG



Tạo một sheet có tên là **NKC**, cấu trúc sổ như sau:

	A	B	C	D	E	F	G
1	SỔ NHẬT KÝ CHUNG						
2	Tháng ... năm ...						
3							
4	Ngày ghi sổ	Chứng từ		Diễn giải	Tài khoản		Số tiền
5		Số	Ngày		Nợ	Có	
6	1	2	3	4	5	6	7
7	[1]	[2]	[3]	[4]	[5]	[6]	[7]
8							
9							
10							
n	Tổng cộng						[8]

Nếu doanh nghiệp có mở các sổ nhật ký chuyên dùng thì các chứng từ có liên quan sẽ được ưu tiên ghi vào các sổ nhật ký chuyên dùng, nếu các chứng từ không phải ghi vào các sổ nhật ký chuyên dùng thì mới ghi sang sổ nhật ký chung.

Các ô có [?] ta phải thiết lập công thức, cụ thể là:

- ✓ **[1] Ngày ghi sổ:**

$$[1] = \text{IF}(\text{OR}(\text{LEFT}(\text{SKTM!}\$H3;3) = "111"; \text{LEFT}(\text{SKTM!}\$I3;3) = "111"; \text{AND}(\text{LEFT}(\text{SKTM!}\$H3;3) = "131"; \text{LEFT}(\text{SKTM!}\$I3;3) <> "131"); \text{AND}(\text{LEFT}(\text{SKTM!}\$I3;3) = "331"; \text{LEFT}(\text{SKTM!}\$H3;3) <> "331")); ""; \text{SKTM!}B3)$$

✓ [2] Số chứng từ:

```
[2] = IF(OR(LEFT(SKTM!$H3;3)="111";  
LEFT(SKTM!$I3;3)="111";AND(LEFT(SKTM!$H3;3)="131";  
LEFT(SKTM!$I3;3)<>"131"); AND(LEFT(SKTM!$I3;3)="331";  
LEFT(SKTM!$H3;3)<>"331"));"";IF(SKTM!D3<>"");  
SKTM!D3;IF(SKTM!C3<>""); SKTM!C3; SKTM!E3)))
```

✓ [3] Ngày chứng từ:

```
[3] = IF(OR(LEFT(SKTM!$H3;3)="111";  
LEFT(SKTM!$I3;3)="111";AND(LEFT(SKTM!$H3;3)="131";  
LEFT(SKTM!$I3;3)<>"131"); AND(LEFT(SKTM!$I3;3)="331";  
LEFT(SKTM!$H3;3)<>"331"));"";SKTM!F3)
```

✓ [4] Diễn giải:

```
[4] = IF(OR(LEFT(SKTM!$H3;3)="111";  
LEFT(SKTM!$I3;3)="111";AND(LEFT(SKTM!$H3;3)="131";  
LEFT(SKTM!$I3;3)<>"131"); AND(LEFT(SKTM!$I3;3)="331";  
LEFT(SKTM!$H3;3)<>"331"));"";SKTM!G3)
```

✓ [5] Tài khoản ghi nợ:

```
[5] = IF(OR(LEFT(SKTM!$H3;3)="111";  
LEFT(SKTM!$I3;3)="111";AND(LEFT(SKTM!$H3;3)="131";  
LEFT(SKTM!$I3;3)<>"131"); AND(LEFT(SKTM!$I3;3)="331";  
LEFT(SKTM!$H3;3)<>"331"));"";SKTM!H3)
```

✓ [6] Tài khoản ghi có:

```
[6] = IF(OR(LEFT(SKTM!$H3;3)="111";  
LEFT(SKTM!$I3;3)="111";AND(LEFT(SKTM!$H3;3)="131";  
LEFT(SKTM!$I3;3)<>"131"); AND(LEFT(SKTM!$I3;3)="331";  
LEFT(SKTM!$H3;3)<>"331"));"";SKTM!I3)
```

- ✓ **[6] Số tiền phát sinh:**

**[6] = IF(OR(LEFT(SKTM!\$H3;3)="111";
LEFT(SKTM!\$I3;3)="111";AND(LEFT(SKTM!\$H3;3)="131";
LEFT(SKTM!\$I3;3)<>"131"); AND(LEFT(SKTM!\$I3;3)="331";
LEFT(SKTM!\$H3;3)<>"331"));"";SKTM!K3)**

- ✓ **[7] Tổng công số tiền phát sinh:**

[7] = SUM(G7: G_{n-1})

Ghi chú: n là dòng thứ n (dòng tổng cộng)

Cuối cùng ta cài đặt bộ lọc tự động AutoFilter vào cột diễn giải để lọc ra những dòng có số liệu.

LẬP SỔ CÁI CÁC TÀI KHOẢN

Để lập sổ cái các tài khoản ta cần phải lập bảng cân đối số phát sinh trước.
Thông tin để lập sổ cái được lấy từ SKTM, BDMTK, BCDSPS.

Tạo sheet mới có tên là **SOCAI**, có cấu trúc như sau:

	A	B	C	D	E	F	G
1	SỔ CÁI						
2	Tháng ... năm ...						
3	Số hiệu:		[1]	[2]			[3]
4							
5	Ngày ghi sổ	Chứng từ		Diễn giải	TK đối ứng	Số tiền phát sinh	
6		Số	Ngày			Nợ	Có
7	Số dư đầu kỳ:					[4]	[5]
8	1	2	3	4	5	6	7
9	[6]	[7]	[8]	[9]	[10]	[11]	[12]
10							
11							
12							
n-1	Tổng công số phát sinh					[13]	[14]
n	Số dư cuối kỳ					[15]	[16]

- ✓ **[1] Số hiệu tài khoản:**

Tại ô C4 nhập vào tài khoản nào thì in ra sổ cái tài khoản của tài khoản đó.

- ✓ [2] Tên tài khoản :

[2]= VLOOKUP(C4;DMTKTH;3;0)

Trong

đó:

DMTKTH là tên vùng ở BCDSPTS.

- ✓ [3] Loại tài khoản (ghi nợ hay có):

[3]= VLOOKUP(C4;BDMTKTH;2;0)

- ✓ [4] Số dư nợ đầu kỳ:

[4]= IF(OR(AND(G4="N";SUMIF(SOHIEUTK;C4&"*";SODDK)>0);
AND(G4="C";SUMIF(SOHIEUTK;C4&"*";SODDK)<0));
ABS(SUMIF(SOHIEUTK;C4&"*";SODDK));0)

- ✓ [5] Số dư có đầu kỳ:

[5]= IF(OR(AND(G4="C";SUMIF(SOHIEUTK;C4&"*";SODDK)>0);
AND(G4="N";SUMIF(SOHIEUTK;C4&"*";SODDK)<0));
ABS(SUMIF(SOHIEUTK;C4&"*";SODDK));0)

- ✓ [6] Ngày ghi sổ:

[6]= IF(OR(LEFT(SKTM!\$H3;3)=\$C\$4;
LEFT(SKTM!\$I3;3)=\$C\$4); SKTM!B3;"")

- ✓ [7] Số chứng từ:

[7]= IF(OR(LEFT(SKTM!\$H3;3)=\$C\$4; LEFT(SKTM!\$I3;3)=\$C\$4);
IF(SKTM!D3<>""; SKTM!D3; IF(SKTM!C3<>"";
SKTM!C3;SKTM!E3));"")

- ✓ [8] Ngày chứng từ:

[8]= IF(OR(LEFT(SKTM!\$H3;3)=\$C\$4;
LEFT(SKTM!\$I3;3)=\$C\$4);SKTM!F3;"")

- ✓ [9] Diễn giải:

[9]= IF(OR(LEFT(SKTM!\$H3;3)=\$C\$4;
LEFT(SKTM!\$I3;3)=\$C\$4);SKTM!G3;"")

- ✓ [10] Tài khoản đối ứng:

[10]= IF(LEFT(SKTM!\$H3;3)=\$C\$4; SKTM!I3;
IF(LEFT(SKTM!\$I3;3)=\$C\$4);SKTM!H3;””))

✓ [11] Số tiền phát sinh nợ:

[11]= IF(LEFT(SKTM!\$H3;3)=\$C\$4; SKTM!K3;0)

✓ [12] Số tiền phát sinh có:

[12]= IF(LEFT(SKTM!\$I3;3)=\$C\$4; SKTM!K3;0)

✓ [13] Tổng số tiền phát sinh nợ:

[13]= SUM(F9:F_{n-2})

✓ [14] Tổng số tiền phát sinh có:

[14]= SUM(G9:G_{n-2})

✓ [15] Số dư nợ cuối kỳ:

[15]= MAX(F7+F_{n-1}- G7-G_{n-1};0)

✓ [16] Số dư có cuối kỳ:


[16]= MAX(G7+G_{n-1}- F7-F_{n-1};0)

Ghi chú: n là dòng đặt dòng số dư cuối kỳ.

Sau khi thiết lập xong các công thức trên ta copy công thức từ [6] đến [12] cho tới dòng n -2. Cài đặt bộ lọc **AutoFilter** để lọc dữ liệu.

LẬP SỔ QUỸ TIỀN MẶT – SỔ TIỀN GỬI NGÂN HÀNG

SỔ QUỸ TIỀN MẶT



ạo sheet mới đặt tên là **SOQTM**, có cấu trúc như sau:

	A	B	C	D	E	F	G	H
1	SỐ QUÝ TIỀN MẶT							
2	Tháng ... năm ...							
3								
4	Ngày	Chứng từ		Diễn giải	TK đối ứng	Số tiền phát sinh		
5	ghi số	Số	Ngày			Thu	Chi	Tồn
6	Tồn quỹ đầu kỳ:							[1]
7	1	2	3	4	5	6	7	8
8	[2]	[3]	[4]	[5]	[6]	[7]	[8]	[9]
9								
10								
11								
n-1	Cộng phát sinh						[10]	[11]
n	Tồn quỹ cuối kỳ							[12]

Ta thiết lập công thức cho các ô [?] như sau:

- ✓ [1] Tồn quỹ đầu kỳ:

[1] = SUMIF(SOHIEUTK;"111*";SODDK)

- ✓ [2] Ngày ghi số:

[2] = IF(OR(LEFT(SKTM!\$H3;3)="111";
LEFT(SKTM!\$I3;3)="111");SKTM!B3;"")

- ✓ [3] Số chứng từ:

[3] = IF(OR(LEFT(SKTM!\$H3;3)="111";
LEFT(SKTM!\$I3;3)="111");SKTM!D3;"")

- ✓ [4] Ngày chứng từ:

[4] = IF(OR(LEFT(SKTM!\$H3;3)="111";
LEFT(SKTM!\$I3;3)="111");SKTM!F3;"")

- ✓ [5] Diễn giải:

[5] = IF(OR(LEFT(SKTM!\$H3;3)="111";
LEFT(SKTM!\$I3;3)="111");SKTM!G3;"")

- ✓ [6] Tài khoản đối ứng:

[6] = IF(LEFT(SKTM!\$H3;3)="111"; SKTM!\$I3;IF(LEFT(SKTM!\$I3;3)="111");SKTM!H3;""))

- ✓ [7] Thu (phát sinh tăng):

[7] = IF(LEFT(SKTM!\$H3;3)="111"; SKTM!\$K3;"")

- ✓ [8] Chi (phát sinh giảm):

[8] = IF(LEFT(SKTM!\$I3;3)="111"; SKTM!\$K3;"")

- ✓ [9] Tồn:

[9] = IF(F8+G8=0;0; \$H\$6+SUM(\$F\$8:\$F8)- SUM(\$G\$8:\$G8))

- ✓ [10] Cộng phát sinh tăng trong kỳ:

[10] = SUM(F8:F_{n-2})

- ✓ [11] Cộng phát sinh giảm trong kỳ:

[11] = SUM(G8:G_{n-2})


- ✓ [12] Tồn quỹ cuối kỳ:

[12] = H6+F_{n-1}-G_{n-1}

Ghi chú: trong các công thức trên, n là dòng thứ n (dòng đặt dòng tồn quỹ cuối kỳ).

Sau khi thiết lập xong các công thức trên ta copy công thức từ [2] đến [9] cho tới dòng n-2. Cài đặt bộ lọc AutoFilter để lọc dữ liệu.

SỔ TIỀN GỬI NGÂN HÀNG



Tạo sheet mới đặt tên là **SOTGNH**, có cấu trúc như sau:

	A	B	C	D	E	F	G	H	
1	SỐ TIỀN GỬI NGÂN HÀNG								
2	Tháng ... năm ...								
3									
4	Ngày ghi sổ	Chứng từ		Diễn giải	TK đối ứng	Số tiền phát sinh			
5		Số	Ngày			Thu	Chi	Tồn	
6	Tồn quỹ đầu kỳ:							[1]	
7	1	2	3	4	5	6	7	8	
8	[2]	[3]	[4]	[5]	[6]	[7]	[8]	[9]	
9									
10									
11									
n-1	Cộng phát sinh						[10]	[11]	
n	Tồn quỹ cuối kỳ								[12]

Ta thiết lập công thức cho các ô [?] như sau:

- ✓ [1] Tồn quỹ đầu kỳ:

[1] = SUMIF(SOHIEUTK;"112";SODDK)

- ✓ [2] Ngày ghi sổ:

[2] = IF(OR(LEFT(SKTM!\$H3;3)="112"; LEFT(SKTM!\$I3;3)="112");SKTM!B3;"")

- ✓ [3] Số chứng từ:

[3] = IF(OR(LEFT(SKTM!\$H3;3)="112"; LEFT(SKTM!\$I3;3)="112");SKTM!D3;"")

- ✓ [4] Ngày chứng từ:

[4] = IF(OR(LEFT(SKTM!\$H3;3)="112"; LEFT(SKTM!\$I3;3)="112");SKTM!F3;"")

- ✓ [5] Diễn giải:

[5] = IF(OR(LEFT(SKTM!\$H3;3)="112"; LEFT(SKTM!\$I3;3)="112");SKTM!G3;"")

- ✓ [6] Tài khoản đối ứng:

[6] = IF(LEFT(SKTM!\$H3;3)="112"; SKTM!\$I3;IF(LEFT(SKTM!\$I3;3)="112");SKTM!H3;""))

- 7] Thu (phát sinh tăng):

[7] = IF(LEFT(SKTM!\$H3;3)="112"; SKTM!\$K3;""))

- ✓ [8] Chi (phát sinh giảm):

[8] = IF(LEFT(SKTM!\$I3;3)="112"; SKTM!\$K3;""))

- ✓ [9] Tồn:

[9] = IF(F8+G8=0;0; \$H\$6+SUM(\$F\$8:\$F8)- SUM(\$G\$8:\$G8))

- ✓ [10] Cộng phát sinh tăng trong kỳ:

[10] = SUM(F8:F_{n-2})

- ✓ [11] Cộng phát sinh giảm trong kỳ:

[11] = SUM(G8:G_{n-2})

- ✓ [12] Tồn quỹ cuối kỳ:


[12] = H6+F_{n-1}-G_{n-1}

Ghi chú: trong các công thức trên, n là dòng thứ n (dòng đặt dòng tồn quỹ cuối kỳ).

Sau khi thiết lập xong các công thức trên ta copy công thức từ [2] đến [9] cho tới dòng n-2. Cài đặt bộ lọc AutoFilter để lọc dữ liệu.

LẬP SỔ CHI TIẾT HÀNG TỒN KHO – BẢNG TỔNG HỢP HÀNG TỒN KHO

SỔ CHI TIẾT HÀNG TỒN KHO



Tạo sheet mới đặt tên là **SCTHTK** , có cấu trúc như sau:

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	SỔ CHI TIẾT HÀNG TỒN KHO												
2	Tháng ... năm...												
3	Số hiệu				[1]	[2]							
4	Ngày	Chứng từ		Diễn giải	TK	Đơn giá N/X	Nhập		Xuất		Tồn		Ghi chú
5	y ghi số	Số	Ngày y		đối ứng		SL	GT	SL	GT	SL	GT	
6	1	2	3	4	5	6	7	8	9	10	11	12	13
7	Tồn đầu kỳ										[3]	[4]	
8	[5]	[6]	[7]	[8]	[9]	[10]	[11]	[12]	[13]	[14]	[15]	[16]	
9													
10													
11													
n-1	Cộng phát sinh						[17]	[18]	[19]	[20]			
n	Tồn cuối kỳ										[21]	[22]	

Ta thiết lập công thức cho các ô [?] như sau:

- ✓ [1] Nhập mã hàng tồn kho (số hiệu tài khoản chi tiết).
- ✓ [2] Tên hàng tồn kho:

[2] = VLOOKUP(\$E\$3;BDMTK;3;0)

- ✓ [3] Tồn đầu kỳ:

[3] = VLOOKUP(\$E\$3;BDMTK;4;0)

- ✓ [4] Giá trị tồn đầu kỳ:

[4] = VLOOKUP(\$E\$3;BDMTK;5;0)

- ✓ [5] Ngày ghi sổ:

[5] = IF(OR(SKTM!\$H3=\$E\$3; SKTM!\$I3=\$E\$3);SKTM!\$B3;"")

- ✓ [6] Số chứng từ:

[6] = IF(OR(SKTM!\$H3=\$E\$3; SKTM!\$I3=\$E\$3);SKTM!\$E3;"")

- ✓ [7] Ngày chứng từ:

[7] = IF(OR(SKTM!\$H3=\$E\$3; SKTM!\$I3=\$E\$3);SKTM!\$F3;"")

- ✓ [8] Diễn giải:

[8] = IF(OR(SKTM!\$H3=\$E\$3; SKTM!\$I3=\$E\$3);SKTM!\$G3;"")

- ✓ [9] Tài khoản đối ứng:

[9] = IF(SKTM!H3=\$E\$3;SKTM!I3;IF(SKTM!I3=\$E\$3; SKTM!H3;""))

- ✓ [10] Đơn giá nhập/xuất:

[10] = IF(G8+I8=0;0;(H8+J8)/(G8+I8))

- ✓ [11] Số lượng nhập:

[11] = IF(SKTM!\$H3=\$E\$3; SKTM!\$J3;0)

- ✓ [12] Giá trị nhập:

[12] = IF(SKTM!\$H3=\$E\$3; SKTM!\$K3;0)

- ✓ [13] Số lượng xuất:

[13] = IF(SKTM!\$I3=\$E\$3; SKTM!\$J3;0)

- ✓ [14] Giá trị xuất:

$$[14] = \text{IF}(\text{SKTM!}\$I3=\$E\$3; \text{SKTM!}\$K3;0)$$

- ✓ [15] Số lượng tồn:

$$[15] = \text{IF}(\text{SUM}(G8:J8)=0;0; \$K\$7+\text{SUM}(\$G\$8:G8)- \text{SUM}(\$I\$8:I8))$$

- ✓ [16] Giá trị tồn:

$$[16] = \text{IF}(\text{SUM}(G8:J8)=0;0; \$L\$7+\text{SUM}(\$H\$8:H8)- \text{SUM}(\$J\$8:J8))$$

- ✓ [17] Tổng số lượng nhập trong kỳ:

$$[17] = (\text{SUM}(G8:G_{n-2}))$$

- ✓ [18] Tổng giá trị nhập trong kỳ:

$$[18] = (\text{SUM}(H8:H_{n-2}))$$

- ✓ [19] Tổng số lượng xuất trong kỳ:

$$[19] = (\text{SUM}(I8:I_{n-2}))$$

- ✓ [20] Tổng giá trị xuất trong kỳ:

$$[20] = (\text{SUM}(J8:J_{n-2}))$$

- ✓ [21] Số lượng tồn cuối kỳ:

$$[21] = K8+G_{n-1}-I_{n-1}$$

- ✓ [22] Giá trị tồn cuối kỳ:


$$[22] = L8+H_{n-1}-J_{n-1}$$

Ghi chú: trong các công thức trên, n là dòng thứ n (dòng đặt dòng tồn cuối kỳ).

Sau khi thiết lập xong các công thức trên ta copy công thức từ [5] đến [16] cho tới dòng n-2. Cài đặt bộ lọc **AutoFilter** để lọc dữ liệu.

LẬP BẢNG TỔNG HỢP CHI TIẾT HÀNG TỒN KHO

- ❖ Lập bảng tổng hợp chi tiết tài khoản 152 (nguyên vật liệu).



Ta tạo sheet mới có tên là **BTHCT152**, có cấu trúc như sau:

	A	B	C	D	E	F	G	H	I	J	K	L
1	BẢNG TỔNG HỢP CHI TIẾT TÀI KHOẢN 152											
2	Tháng ... năm...											
3	Mã hàng	Tên hàng	Đơn vị tính	Tồn đầu kỳ		Nhập		Xuất		Tồn cuối kỳ		
4				SL	GT	SL	GT	SL	GT	SL	GT	ĐG
5		[1]		[2]	[3]	[4]	[5]	[6]	[7]	[8]	[9]	[10]
6												
7												
8												
9												
10												
11												
n-1												
n	Tổng cộng				[11]		[12]		[13]			

Các ô [?] ta thiết lập công thức như sau:

- ✓ [1] Tên vật liệu:

$$[1] = \text{VLOOKUP}(A5; \text{BDMTK}; 3; 0)$$

- ✓ [2] Số lượng tồn đầu kỳ:

$$[2] = \text{VLOOKUP}(A5; \text{BDMTK}; 4; 0)$$

- ✓ [3] Giá trị tồn đầu kỳ:

$$[3] = \text{VLOOKUP}(A5;BDMTK;5;0)$$

- ✓ [4] Số lượng nhập trong kỳ:

$$[4] = \text{SUMIF}(\text{TKGHINO};A5;\text{SOLUONGPS})$$

- ✓ [5] Giá trị nhập trong kỳ:

$$[5] = \text{SUMIF}(\text{TKGHINO};A5;\text{SOTIENPS})$$

- ✓ [6] Số lượng xuất trong kỳ:

$$[6] = \text{SUMIF}(\text{TKGHICO};A5;\text{SOLUONGPS})$$

- ✓ [7] Giá trị xuất trong kỳ:

$$[7] = \text{SUMIF}(\text{TKGHICO};A5;\text{SOTIENPS})$$

- ✓ [8] Số lượng tồn cuối kỳ:

$$[8] = D5+F5-H5$$

- ✓ [9] Giá trị tồn cuối kỳ:

$$[9] = E5+G5-I5$$

- ✓ [10] Đơn giá tồn cuối kỳ:

$$[10] = K5/J5$$

Ghi chú: trong các công thức trên, n là dòng thứ n (dòng đặt dòng tổng cộng). Sau khi thiết lập xong các công thức trên ta copy công thức từ [1] đến [10] cho tới dòng n-1.

✓ [11] Tổng giá trị tồn đầu kỳ:

$$[11]=SUM(E5:E_{n-1})$$

✓ [12] Tổng giá trị nhập trong kỳ:


$$[12]=SUM(G5:G_{n-1})$$

✓ [13] Tổng giá trị xuất trong kỳ:

$$[13]=SUM(I5:I_{n-1})$$

Lưu ý: Lập bảng tổng hợp chi tiết tài khoản 153, 155, 156 được lập tương tự như bảng tổng hợp chi tiết tài khoản 152.

SỔ CHI TIẾT CÔNG NỢ VÀ SỔ CHI TIẾT CÁC TÀI KHOẢN KHÁC



Ta tạo sheet mới đặt tên là **SOCTTK** có cấu trúc như sau:

	A	B	C	D	E	F	G	H	I	
1	[1]									
2	Tháng .. năm ...									
3	Số hiệu [2]									
4	[3]									
5	Ngày ghi	Chứng từ		Diễn	TK đối	Số phát sinh		Số dư		
6	số	Số	ngày	giải	ứng	Nợ	Có	Nợ	Có	
7	1	2	3	4	5	6	7	8	9	
8	Số dư đầu kỳ							[4]	[5]	
9	[6]	[7]	[8]	[9]	[10]	[11]	[12]	[13]	[14]	
10										
11										
n-1	Tổng cộng số phát sinh						[15]	[16]		
N	Số dư cuối kỳ							[17]	[18]	

Các ô [?] ta thiết lập công thức như sau:

✓ [1] Tiêu đề:

```
[1] = IF(LEFT(E3;3)="131"; "SỐ CHI TIẾT PHẢI THU KHÁCH HÀNG";  
IF(LEFT(E3;3)="331"; "SỐ CHI TIẾT PHẢI TRẢ NGƯỜI BÁN"; "SỐ  
CHI TIẾT CÁC TÀI KHOẢN KHÁC"))
```

✓ [2] Số hiệu tài khoản (mã khách hàng):

✓ [3] Tên khách hàng:

```
[3] = "Tên tài khoản:  
&VLOOKUP(E3;BDMTK;3;0)
```

✓ [4] Số dư nợ đầu kỳ:

```
[4] = IF(VLOOKUP($E3;BDMTK;2;0)="N";  
MAX(0;VLOOKUP($E3;BDMTK;5;0));  
ABS(MIN(0;VLOOKUP($E3;BDMTK;5;0))))
```

✓ [5] Số dư có đầu kỳ:

```
[5] = IF(VLOOKUP($E3;BDMTK;2;0)="C";  
MAX(0;VLOOKUP($E3;BDMTK;5;0));  
ABS(MIN(0;VLOOKUP($E3;BDMTK;5;0))))
```

✓ [6] Ngày ghi sổ:

```
[6] = IF(OR(SKTM!$H3=$E$3; SKTM!$I3=$E$3);SKTM!B3;"")
```

✓ [7] Số chứng từ:

```
[7] = IF(OR(SKTM!$H3=$E$3;  
SKTM!$I3=$E$3);IF(SKTM!D3<>"";SKTM!D3;if(  
SKTM!C3<>"";SKTM!C3;SKTM!C3));"")
```

✓ [8] Số chứng từ:

```
[8] = IF(OR(SKTM!$H3=$E$3; SKTM!$I3=$E$3);SKTM!F3;"")
```

✓ [9] Diễn giải:

```
[9] = IF(OR(SKTM!$H3=$E$3; SKTM!$I3=$E$3);SKTM!G3;"")
```

✓ [10] Tài khoản đối ứng :

```
[10] = IF(SKTM!$H3=$E$3;SKTM!$I3;IF(SKTM!$I3=$E$3;SKTM!H3;""))
```

- ✓ [11] Số phát sinh nợ :

$$[11] = \text{IF}(\text{SKTM!H3}=\text{\$E\$3};\text{SKTM!K3};0)$$

- 12] Số phát sinh có :

$$[12] = \text{IF}(\text{SKTM!I3}=\text{\$E\$3};\text{SKTM!K3};0)$$

- ✓ [13] Số dư nợ:

$$[13] = \text{IF}(\text{F9}+\text{G9}=0;0;\text{MAX}(\text{\$H\$8}+\text{SUM}(\text{\$F\$9:F9})-\text{SUM}(\text{\$G\$9:G9})-\text{\$I\$8};0))$$

- ✓ [14] Số dư có:

$$[14] = \text{IF}(\text{F9}+\text{G9}=0;0;\text{MAX}(\text{\$I\$8}+\text{SUM}(\text{\$G\$9:G9})-\text{SUM}(\text{\$F\$9:F9})-\text{\$H\$8};0))$$

- ✓ [15] Tổng phát sinh nợ trong kỳ:

$$[15] = \text{SUM}(\text{F9:F}_{n-2})$$

- ✓ [16] Tổng phát sinh có trong kỳ:

$$[16] = \text{SUM}(\text{G9:G}_{n-2})$$

- ✓ [17] Số dư nợ cuối kỳ:


$$[17] = \text{MAX}(0;\text{H8}+\text{F}_{n-1}-\text{I8}-\text{G}_{n-1})$$

- ✓ [18] Số dư có cuối kỳ:

$$[18] = \text{MAX}(0;\text{I8}+\text{G}_{n-1}-\text{H8}-\text{F}_{n-1})$$

Ghi chú: trong các công thức trên n là dòng thứ n (dòng đặt dòng số dư cuối kỳ). Sau khi thiết lập công thức xong ta copy công thức từ ô [6] đến ô [14] xuống dòng n-2 và cài bộ lọc tự động AutoFilter.

BẢNG TỔNG HỢP CHI TIẾT PHẢI THU KHÁCH HÀNG



Ta tạo sheet mới đặt tên là **THCT131**

	A	B	C	D	E	F	G	H
1	BẢNG TỔNG HỢP CHI TIẾT TK 131							
2	Tháng ... năm...							
3								
4	Mã khách	Tên khách	Số dư đầu kỳ		Số phát sinh		Số dư cuối kỳ	
5	hàng	hàng	Nợ	Có	Nợ	Có	Nợ	Có
6			[1]	[2]	[3]	[4]	[5]	[6]
7								
8								
9								
10								
11								
12								
...								
	Tổng cộng							

Cột mã khách hàng và tên khách hàng chính là tên và số hiệu tài khoản chi tiết phải thu khách hàng. Ta thiết lập công thức cho các ô [?] như sau:

- ✓ [1] Số dư nợ đầu kỳ:

$$[1] = \text{MAX}(0; \text{VLOOKUP}(A6; \text{BDMTK}; 5; 0))$$

- ✓ [2] Số dư có đầu kỳ:

$$[2] = \text{ABS}(\text{MIN}(0; \text{VLOOKUP}(A6; \text{BDMTK}; 5; 0)))$$

- ✓ [3] Số phát sinh nợ trong kỳ:

$$[3] = \text{SUMIF}(\text{TKGHINO}; A6; \text{SOTIENPS})$$

✓ [4] Số phát sinh có trong kỳ:

[4] = SUMIF(TKGHICO;A6;SOTIENPS)

✓ [5] Số dư nợ cuối kỳ:


[5] = MAX(0;C6+E6-D6-F6)

✓ [6] Số dư có cuối kỳ:

[6] = MAX(0;D6+F6-C6-E6)

Sau khi thiết lập xong công thức từ ô [1] tới ô [6] ta copy xuống hết bảng và dùng hàm Sum() để cộng số tiền cho dòng tổng cộng.

BẢNG TỔNG HỢP CHI TIẾT PHẢI TRẢ NGƯỜI BÁN



Ta tạo sheet mới đặt tên là THCT331

	A	B	C	D	E	F	G	H
1	BẢNG TỔNG HỢP CHI TIẾT TK 331							
2	Tháng ... năm...							
3								
4	Mã nhà	Tên nhà	Số dư đầu kỳ		Số phát sinh		Số dư cuối kỳ	
5	CC	cung cấp	Nợ	Có	Nợ	Có	Nợ	Có
6			[1]	[2]	[3]	[4]	[5]	[6]
7								
8								
9								
10								
...								
	Tổng cộng							

Cột mã nhà cung cấp và tên nhà cung cấp chính là tên và số hiệu tài khoản chi tiết phải trả nhà cung cấp. Ta thiết lập công thức cho các ô [?] như sau:

✓ [1] Số dư nợ đầu kỳ:

[1] =ABS(MIN(0;VLOOKUP(A6;BDMTK;5;0)))

✓ [2] Số dư có đầu kỳ:

[2] = MAX(0;VLOOKUP(A6;BDMTK;5;0))

✓ [3] Số phát sinh nợ trong kỳ:

[3] = SUMIF(TKGHINO;A6;SOTIENPS)

✓ [4] Số phát sinh có trong kỳ:

[4] = SUMIF(TKGHICO;A6;SOTIENPS)

✓ [5] Số dư nợ cuối kỳ:

[5] = MAX(0;C6+E6-D6-F6)

✓ [6] Số dư có cuối kỳ:

[6] = MAX(0;D6+F6-C6-E6)

Sau khi thiết lập xong công thức từ ô [1] tới ô [6] ta copy xuống hết bảng và dùng hàm Sum() để cộng số tiền cho dòng tổng cộng.

LẬP BẢNG CÂN ĐỐI SỐ PHÁT SINH (BẢNG CÂN ĐỐI TÀI KHOẢN)

Ta tạo sheet mới đặt tên là BCDSPS, có cấu trúc như sau:

	A	B	C	D	E	F	G	H	I
1	BẢNG CÂN ĐỐI SỐ PHÁT SINH								
2	Tháng ... năm ...								
3									
4	Mã	Loại	Tên tài khoản	Số dư đầu kỳ		Số phát sinh		Số dư cuối kỳ	
5	TK	TK		Nợ	Có	Nợ	Có	Nợ	Có
6	111	N	Tiền mặt	[1]	[2]	[3]	[4]	[5]	[6]
7	112	N	Tiền gửi Ngân hàng						
8	113	N	Tiền đang chuyển						
9	121	N	Đầu tư chứng khoán ngắn hạn						
10	128	N	Đầu tư ngắn hạn khác						
11	129	C	Dự phòng giảm giá đầu tư ngắn hạn						
12	131	N	Phải thu của khách hàng						
13	133	N	Thuế GTGT được khấu trừ						
14	136	N	Phải thu nội bộ						
15	138	N	Phải thu khác						
16	139	C	Dự phòng phải thu khó đòi						
17	141	N	Tạm ứng						
18	142	N	Chi phí trả trước ngắn hạn						
19	144	N	Cầm cố, ký quỹ, ký cược ngắn hạn						
20	151	N	Hàng mua đang đi đường						
21	152	N	Nguyên liệu, vật liệu						
22	153	N	Công cụ, dụng cụ						
23	154	N	Chi phí sản xuất, kinh doanh dở dang						
24	155	N	Thành phẩm						
25	156	N	Hàng hóa						
26	157	N	Hàng gửi đi bán						
27	158	N	Hàng hoá kho bảo thuế						
28	159	C	Dự phòng giảm giá hàng tồn kho						
29	161	N	Chi sự nghiệp						
30	211	N	Tài sản cố định hữu hình						
31	212	N	Tài sản cố định thuê tài chính						
32	213	N	Tài sản cố định vô hình						
33	214	C	Hao mòn tài sản cố định						
34	217	N	Bất động sản đầu tư						
35	221	N	Đầu tư vào công ty con						
36	222	N	Vốn góp liên doanh						
37	223	N	Đầu tư vào công ty liên kết						
38	228	N	Đầu tư dài hạn khác						
39	229	C	Dự phòng giảm giá đầu tư dài hạn						
40	241	N	Xây dựng cơ bản dở dang						
41	242	N	Chi phí trả trước dài hạn						
42	243	N	Tài sản thuế thu nhập hoãn lại						

43	244	N	Ký quỹ, ký cược dài hạn						
44	311	C	Vay ngắn hạn						
45	315	C	Nợ dài hạn đến hạn trả						
46	331	C	Phải trả cho người bán						
47	333	C	Thuế và các khoản phải nộp NN						
48	334	C	Phải trả người lao động						
49	335	C	Chi phí phải trả						
50	336	C	Phải trả nội bộ						
51	337	C	Thanh toán theo tiến độ kế hoạch hợp đồng xây dựng						
52	338	C	Phải trả, phải nộp khác						
53	341	C	Vay dài hạn						
54	342	C	Nợ dài hạn						
55	343	C	Trái phiếu phát hành						
56	344	C	Nhận ký quỹ, ký cược dài hạn						
57	347	C	Thuế thu nhập hoãn lại phải trả						
58	351	C	Quỹ dự phòng trợ cấp mất việc làm						
59	352	C	Dự phòng phải trả						
60	411	C	Nguồn vốn kinh doanh						
61	412	C	Chênh lệch đánh giá lại tài sản						
62	413	C	Chênh lệch tỷ giá hối đoái						
63	414	C	Quỹ đầu tư phát triển						
64	415	C	Quỹ dự phòng tài chính						
65	418	C	Các quỹ khác thuộc vốn chủ sở hữu						
66	419	C	Cổ phiếu quỹ						
67	421	C	Lợi nhuận chưa phân phối						
68	431	C	Quỹ khen thưởng, phúc lợi						
69	441	C	Nguồn vốn đầu tư xây dựng cơ bản						
70	461	C	Nguồn kinh phí sự nghiệp						
71	466	C	Nguồn kinh phí đã hình thành TSCĐ						
72	511	C	Doanh thu bán hàng và cung cấp DV						
73	512	C	Doanh thu bán hàng nội bộ						
74	515	C	Doanh thu hoạt động tài chính						
75	521	N	Chiết khấu thương mại						
76	531	N	Hàng bán bị trả lại						
77	532	N	Giảm giá hàng bán						
78	611	N	Mua hàng						
79	621	N	Chi phí nguyên liệu, vật liệu trực tiếp						
80	622	N	Chi phí nhân công trực tiếp						
81	623	N	Chi phí sử dụng máy thi công						
82	627	N	Chi phí sản xuất chung						
83	631	N	Giá thành sản xuất						
84	632	N	Giá vốn hàng bán						
85	635	N	Chi phí tài chính						
86	641	N	Chi phí bán hàng						
87	642	N	Chi phí quản lý doanh nghiệp						

88	711	C	Thu nhập khác						
89	811	N	Chi phí khác						
90	821	N	Chi phí thuế thu nhập doanh nghiệp						
91	911	N	Xác định kết quả kinh doanh						

Các ô [?] ta thiết lập công thức như sau:

✓ **[1] Số dư nợ đầu kỳ:**

Chỉ tiêu này tổng hợp từ số dư đầu kỳ trên các tài khoản chi tiết trong bảng danh mục tài khoản. Như các bạn đã biết trong **BDMTK** tài khoản nào có loại tài khoản là “N” mà số dư đầu kỳ âm thì số dư đó là số dư có và tài khoản nào có loại tài khoản là “C” mà số dư đầu kỳ âm thì số dư đó là số dư nợ, cho nên khi xử lý số dư để ghi vào bảng cân đối số phát sinh ta phải chuyển số dư đó sang cột đối diện.

[1]=IF(OR(AND(B6="N";SUMIF(SOHIEUTK;A6&"*";SODDK)>0);AND(B6="C";SUMIF(SOHIEUTK;A6&"*";SODDK)<0));ABS(SUMIF(SOHIEUTK;A6&"*";SODDK));0)

✓ **[2] Số dư có đầu kỳ:**

[2]=IF(OR(AND(B6="N";SUMIF(SOHIEUTK;A6&"*";SODDK)<0);AND(B6="C";SUMIF(SOHIEUTK;A6&"*";SODDK)>0));ABS(SUMIF(SOHIEUTK;A6&"*";SODDK));0)

✓ **[3] Số phát sinh nợ trong kỳ:**

[3]=SUMIF(TKGHINO;A6&"*";SOTIENPS)

✓ **[4] Số phát sinh có trong kỳ:**

[4]=SUMIF(TKGHICO;A6&"*";SOTIENPS)

✓ **[5] Số nợ cuối kỳ:**

[5]=MAX(0;D6+F6-E6-G6)

✓ **[6] Số có cuối kỳ:**

[6]=MAX(0;E6+G6-D6-F6)

Sau khi thiết lập xong công thức trên ta copy từ ô **[1]** tới ô **[6]** xuống tài khoản cuối cùng.

LẬP BÁO CÁO TÀI CHÍNH
LẬP BẢNG CÂN ĐỐI KẾ TOÁN

- **Trở lại bảng danh mục tài khoản (BDMTK)**

Như đã đề cập ở phần đầu thì mỗi tài khoản được thiết lập một mã TS-NV tương ứng với từng chỉ tiêu trong bảng cân đối kế toán, nhưng mã TS-NV thiết lập ban đầu chỉ phù hợp với tính chất cơ bản của các tài khoản mà thôi. Trong thực tế số dư của một số tài khoản thường xuyên biến đổi từ dư nợ sang dư có và ngược lại. Do đó tại thời điểm lập báo cáo (cuối kỳ) ta cần phải xem xét số dư của từng tài khoản còn phù hợp với tính chất cơ bản của tài khoản đó nữa hay không. Nếu còn phù hợp thì mã TS_NV vẫn giữ nguyên mã cũ, ngược lại thì phải đổi thành mã TS_NV mới cho phù hợp với số dư cuối kỳ tại thời điểm lập báo cáo. Để làm được điều này ta cần thực hiện thiết lập lại mã TS_NV cho phù hợp với số dư cuối kỳ các tài khoản. Xem xét trong hệ thống tài khoản và theo quy định hiện hành chỉ có các tài khoản: 131, 138, 141, 331, 333, 337, 338 khi lập bảng cân đối kế toán, số dư cuối kỳ của nó không thể bù trừ cho nhau giữa dư nợ và dư có được mà phải tổng hợp hai loại số dư này riêng để ghi vào hai chỉ tiêu khác nhau trong bảng cân đối kế toán, ví dụ như:

+ Đối với tài khoản 131: số dư nợ ghi vào chỉ tiêu “Phải thu của khách hàng” ứng với mã TS-NV thiết lập ban đầu là “100.130.131”, ngược lại số dư có ghi vào chỉ tiêu “Người mua trả tiền trước” ứng với mã TS-NV “300.310.313”. Như vậy tất cả những tài khoản chi tiết của 131 có số dư bên nợ thì mã cấp không thay đổi, nhưng tất cả những tài khoản chi tiết của 131 có số dư bên có thì mã TS-NV phải đổi thành “300.310.313” để phù hợp với số dư tại thời điểm lập báo cáo.

+ Tương tự đối với tài khoản 331: số dư có ghi vào chỉ tiêu “Phải trả người bán” có mã TS_NV đã thiết lập tương ứng là “300.310.312”, số dư nợ ghi vào chỉ tiêu “Trả trước cho người bán” ứng với mã TS_NV “100.130.132”. Như vậy tất cả những tài khoản chi tiết của tài khoản 331 có số dư bên có thì mã TS_NV không đổi, nhưng tất cả những tài khoản chi tiết của tài khoản 331 có số dư bên nợ thì mã TS_NV phải đổi thành “100.130.132” cho phù hợp với số dư cuối kỳ.

+ Tương tự đối với tài khoản 338: số dư có ghi vào chỉ tiêu “Phải trả khác” có mã TS_NV tương ứng là “300.310.319”, số dư nợ ghi vào chỉ tiêu “Phải thu khác” ứng với mã TS_NV “100.130.138”. Như vậy tất cả những tài khoản chi tiết của 338 có số dư bên có thì mã TS_NV không thay đổi, nhưng những tài khoản chi tiết của 338 có số dư bên nợ thì mã TS_NV phải đổi thành “100.130.138” cho phù hợp với số dư cuối kỳ.

+ Tương tự đối với tài khoản 138 và 141: số dư nợ ghi vào chỉ tiêu “Phải thu khác” có mã TS_NV đã thiết lập tương ứng là “100.130.138”, số dư có ghi vào chỉ tiêu “Phải trả khác” ứng với mã TS_NV “300.310.319”. Như vậy tất cả những tài khoản chi tiết của 138, 141 có số dư bên nợ thì mã TS_NV không thay đổi, nhưng những tài khoản chi tiết của 138, 141 có số dư bên có thì mã TS_NV phải đổi thành “300.310.319” cho phù hợp với số dư cuối kỳ.

+ Tương tự đối với tài khoản 337: số dư có ghi vào chỉ tiêu “Phải trả theo tiến độ kế hoạch hợp đồng xây dựng” có mã TS_NV đã thiết lập tương ứng là “300.310.318”, số dư nợ ghi vào chỉ tiêu “Phải thu theo tiến độ kế hoạch hợp đồng xây dựng” ứng với mã TS_NV “100.130.134”. Như vậy tất cả những tài khoản chi tiết của 337 có số dư bên có thì mã TS_NV không thay đổi, nhưng những tài khoản chi tiết của 337 có số dư bên nợ thì mã TS_NV phải đổi thành “100.130.134” cho phù hợp với số dư cuối kỳ.

Để nhận biết được các tài khoản trên có số dư ngược với tính chất của nó hay không ta thêm một cột vào bảng danh mục tài khoản (**BDMTK**) có tên cột là: mã “TS_NV điều chỉnh” và sử dụng công thức sau:

```
=IF(AND(LEFT(B3;3)="131"; J3<0);"300-310-313";
IF(AND(LEFT(B3;3)="331";J3<0);"100-130-132";
IF(AND(OR(LEFT(B3;3)="138";LEFT(B3;3)="144");J3<0);"300-310-319";
IF(AND(LEFT(B3;3)="338";J3<0);"100-130-135";
IF(AND(LEFT(B3;3)="337";J3<0);"100-130-134";
IF(AND(LEFT(B3;3)="333";J3<0);"100-150-154";A3))))))
```

```
=IF(OR(AND(LEFT(B3;3)="131";J3<0);AND(LEFT(B3;3)="331";J3<0);
AND(OR(LEFT(B3;3)="138";LEFT(B3;3)="141");J3<0);
AND(LEFT(B3;3)="338";J3<0);AND(LEFT(B3;3)="337";J3<0);
AND(LEFT(B3;3)="333";J3<0); LEFT(B3;3)="214"; MID(B3;3;1)="9");
-1;1)*J3
```

Để xử lý số dư cuối kỳ cho phù hợp với từng chỉ tiêu trong bảng cân đối kế toán như lập luận ở trên, những tài khoản chi tiết của 131, 141, 138 có số dư có và những tài khoản chi tiết 331, 333, 337, 338 có số dư nợ thì số dư đang là số âm, để lập được bảng cân đối kế toán ta phải chuyển thành số dương. Những tài khoản thuộc loại “Dự phòng” và “hao mòn” thì số dư của nó đang là số dương phải chuyển sang thành số âm trước khi lập bảng cân đối kế toán. Để giải quyết vấn đề này ta thêm một cột có tên là Số dư cuối kỳ điều chỉnh(SDCKDC) trong bảng danh mục tài khoản (BDMTK) và thiết lập công thức như sau:

Ta đặt tên hai cột vừa thiết lập công thức trong **BDMTK** có tên lần lượt là: **MATSNVDC, SODUCKDC**.

Tạo sheet mới đặt tên là **BCDKT** có cấu trúc như sau:

Đơn vị báo cáo:.....

Địa chỉ:.....

Mẫu số B 01 – DN

(Ban hành theo QĐ số 15/2006/QĐ-BTC

Ngày 20/03/2006 của Bộ trưởng BTC)

BẢNG CÂN ĐỐI KẾ TOÁN

Ngày ... tháng ... năm ...

Đơn vị tính:.....

TÀI SẢN	Mã số	Số cuối năm	Số đầu năm
1	2	4	5
A –Tài sản ngắn hạn (100=110+120+130+140+150)	100	[1]	[2]
I. Tiền và các khoản tương đương tiền	110		
1. Tiền	111		
2. Các khoản tương đương tiền	112		
II. Các khoản đầu tư tài chính ngắn hạn	120		
1. Đầu tư ngắn hạn	121		
2. Dự phòng giảm giá đầu tư ngắn hạn (*)	129		
III. Các khoản phải thu ngắn hạn	130		
1. Phải thu khách hàng	131		
2. Trả trước cho người bán	132		
3. Phải thu nội bộ ngắn hạn	133		
4. Phải thu theo tiến độ kế hoạch hợp đồng xây dựng	134		
5. Các khoản phải thu khác	135		
6. Dự phòng phải thu ngắn hạn khó đòi (*)	139		
IV. Hàng tồn kho	140		
1. Hàng tồn kho	141		
2. Dự phòng giảm giá hàng tồn kho (*)	149		
V. Tài sản ngắn hạn khác	150		
1. Chi phí trả trước ngắn hạn	151		
2. Thuế GTGT được khấu trừ	152		
3. Thuế và các khoản khác phải thu Nhà nước	154		
5. Tài sản ngắn hạn khác	158		
B - TÀI SẢN DÀI HẠN (200 = 210 + 220 + 240 + 250 + 260)	200		
I- Các khoản phải thu dài hạn	210		
1. Phải thu dài hạn của khách hàng	211		
2. Vốn kinh doanh ở đơn vị trực thuộc	212		
3. Phải thu dài hạn nội bộ	213		
4. Phải thu dài hạn khác	218		
5. Dự phòng phải thu dài hạn khó đòi (*)	219		

II. Tài sản cố định	220		
1. Tài sản cố định hữu hình	221		
- Nguyên giá	222		
- Giá trị hao mòn lũy kế (*)	223		
2. Tài sản cố định thuê tài chính	224		
- Nguyên giá	225		
- Giá trị hao mòn lũy kế (*)	226		
3. Tài sản cố định vô hình	227		
- Nguyên giá	228		
- Giá trị hao mòn lũy kế (*)	229		
4. Chi phí xây dựng cơ bản dở dang	230		
III. Bất động sản đầu tư	240		
- Nguyên giá	241		
- Giá trị hao mòn lũy kế (*)	242		
IV. Các khoản đầu tư tài chính dài hạn	250		
1. Đầu tư vào công ty con	251		
2. Đầu tư vào công ty liên kết, liên doanh	252		
3. Đầu tư dài hạn khác	258		
4. Dự phòng giảm giá đầu tư tài chính dài hạn (*)	259		
V. Tài sản dài hạn khác	260		
1. Chi phí trả trước dài hạn	261		
2. Tài sản thuế thu nhập hoãn lại	262		
3. Tài sản dài hạn khác	268		
Tổng cộng tài sản (270 = 100 + 200)	270		
NGUỒN VỐN			
A – Nợ phải trả (300 = 310 + 330)	300		
I. Nợ ngắn hạn	310		
1. Vay và nợ ngắn hạn	311		
2. Phải trả người bán	312		
3. Người mua trả tiền trước	313		
4. Thuế và các khoản phải nộp Nhà nước	314		
5. Phải trả người lao động	315		
6. Chi phí phải trả	316		
7. Phải trả nội bộ	317		
8. Phải trả theo tiến độ kế hoạch hợp đồng xây dựng	318		
9. Các khoản phải trả, phải nộp ngắn hạn khác	319		
10. Dự phòng phải trả ngắn hạn	320		
II. Nợ dài hạn	330		

1. Phải trả dài hạn người bán	331		
2. Phải trả dài hạn nội bộ	332		
3. Phải trả dài hạn khác	333		
4. Vay và nợ dài hạn	334		
5. Thuế thu nhập hoãn lại phải trả	335		
6. Dự phòng trợ cấp mất việc làm	336		
7. Dự phòng phải trả dài hạn	337		
B - VỐN CHỦ SỞ HỮU (400 = 410 + 430)	400		
I. Vốn chủ sở hữu	410		
1. Vốn đầu tư của chủ sở hữu	411		
2. Thặng dư vốn cổ phần	412		
3. Vốn khác của chủ sở hữu	413		
4. Cổ phiếu quỹ (*)	414		
5. Chênh lệch đánh giá lại tài sản	415		
6. Chênh lệch tỷ giá hối đoái	416		
7. Quỹ đầu tư phát triển	417		
8. Quỹ dự phòng tài chính	418		
9. Quỹ khác thuộc vốn chủ sở hữu	419		
10. Lợi nhuận sau thuế chưa phân phối	420		
11. Nguồn vốn đầu tư XDCB	421		
II. Nguồn kinh phí và quỹ khác	430		
1. Quỹ khen thưởng, phúc lợi	431		
2. Nguồn kinh phí	432		
3. Nguồn kinh phí đã hình thành TSCĐ	433		
Tổng cộng nguồn vốn (440 = 300+400)	440		

CÁC CHỈ TIÊU NGOÀI BẢNG CÂN ĐỐI KẾ TOÁN

CHỈ TIÊU	Thuyết minh	Số cuối năm (3)	Số đầu năm (3)
1. Tài sản thuê ngoài			
2. Vật tư, hàng hóa nhận giữ hộ, nhận gia công			
3. Hàng hóa nhận bán hộ, nhận ký gửi, ký cược			
4. Nợ khó đòi đã xử lý			
5. Ngoại tệ các loại			
6. Dự toán chi sự nghiệp, dự án			

Ta thiết lập công thức như sau:

✓ [1] số dư đầu năm: Lấy số dư cuối kỳ của bảng cân đối kế toán năm trước - gõ vào từ bàn phím.

✓ [2] Số dư cuối kỳ:

[2]= SUMIF(MTSNVDC;"*"&C10&"*";SDCKDC)

Trong công thức trên ô C10 là ô chứa mã số của chỉ tiêu đầu tiên trong bảng cân đối kế toán. Copy công thức [2] xuống hết bảng cân đối kế toán, riêng chỉ tiêu tổng tài sản = mã số 100 + mã số 200; chỉ tiêu tổng nguồn vốn = mã số 300 + mã số 400.

BÁO CÁO KẾT QUẢ HOẠT ĐỘNG KINH DOANH



Tạo sheet mới đặt tên là **KQKD** có cấu trúc như sau:

Đơn vị báo cáo:

Mẫu số B 02 – DN

Địa chỉ:.....

(Ban hành theo QĐ số 15/2006/QĐ-BTC

Ngày 20/03/2006 của Bộ trưởng BTC)

BÁO CÁO KẾT QUẢ HOẠT ĐỘNG KINH DOANH

Năm.....

Đơn vị tính:.....

CHỈ TIÊU	Mã số	Năm nay	Năm trước
1	2	4	5
1. Doanh thu bán hàng và cung cấp dịch vụ	01	[1]	
2. Các khoản giảm trừ doanh thu	02	[2]	
3. Doanh thu thuần về bán hàng và cung cấp dịch vụ (10 = 01 - 02)	10	[3]	
4. Giá vốn hàng bán	11	[4]	
5. Lợi nhuận gộp về bán hàng và cung cấp dịch vụ (20 = 10 - 11)	20	[5]	
6. Doanh thu hoạt động tài chính	21	[6]	
7. Chi phí tài chính	22	[7]	
- Trong đó: Chi phí lãi vay	23		
8. Chi phí bán hàng	24	[8]	
9. Chi phí quản lý doanh nghiệp	25	[9]	
10 Lợi nhuận thuần từ hoạt động kinh doanh {30 = 20 + (21 - 22) - (24 + 25)}	30	[10]	
11. Thu nhập khác	31	[11]	
12. Chi phí khác	32	[12]	
13. Lợi nhuận khác (40 = 31 - 32)	40	[13]	

14. Tổng lợi nhuận kế toán trước thuế (50 = 30 + 40)	50	[14]	
15. Chi phí thuế TNDN hiện hành	51	[15]	
16. Chi phí thuế TNDN hoãn lại	52	[16]	
17. Lợi nhuận sau thuế thu nhập doanh nghiệp (60 = 50 – 51 - 52)	60	[17]	

Lập, ngày ... tháng ... năm ...

Người lập biểu

(Ký, họ tên)

Kế toán trưởng

(Ký, họ tên)

Giám đốc

(Ký, họ tên, đóng dấu)

Ghi chú: (*) Chỉ tiêu này chỉ áp dụng đối với công ty cổ phần.

- ❖ Cột năm trước ta nhập từ bản phím số phát sinh năm trước.
- ❖ Cột năm nay: ta thiết lập công thức cho các ô [?] như sau:
- ✓ **[1] Doanh thu bán hàng và cung cấp dịch vụ:**

Số tiền để ghi vào chỉ tiêu này là lấy từ tổng phát sinh bên có các TK 511 và 512 trong kỳ. Lập vùng điều kiện và đặt tên là DKDT như sau:

TKGHICO
511*
512*

[1]= DSUM(SKTM;11;DKDT)

- ✓ **[2] Các khoản giảm trừ doanh thu.**

Số liệu để ghi vào chỉ tiêu này căn cứ vào số phát sinh bên có các tài khoản 521, 531, 532, 3331, 3332, 3333 đối ứng với nợ tài khoản 511, 512. Ta lập vùng điều kiện và đặt tên là DKGTDT.

TKGHINO	TKGHICO
511*	3332*
512*	3332*
511*	3333*
512*	3333*
511*	3331*
512*	3331*
511*	521*

512*	521*
511*	531*
512*	531*
511*	532*
512*	532*

[2]= DSUM(SKTM;11;DKGTDT)

- ✓ **[3] Doanh thu thuần về bán hàng và cung cấp dịch:**

[3]= [1] – [2]

- ✓ **[4] Giá vốn hàng bán:**

Số liệu để ghi vào chỉ tiêu này là số tiền phát sinh bên có tài khoản 632 đối ứng với nợ tài khoản 911. Ta lập vùng điều kiện và đặt tên là DKGV.

TKGHINO	TKGHICO
911*	632*

[4]= DSUM(SKTM;11;DKGV)

- ✓ **[5] Lợi nhuận gộp về bán hàng và cung cấp dịch vụ.**

[5]= [3] – [4]

- ✓ **[6] Doanh thu hoạt động tài chính.**

Số liệu để ghi vào chỉ tiêu này là lấy từ số phát sinh bên nợ tài khoản 515 đối ứng với có tài khoản 911 trong kỳ. Ta lập vùng điều kiện và đặt tên là DKDTTC:

TKGHINO	TKGHICO
515*	911*

[6]= DSUM(SKTM;11;DKDTTC)

- ✓ **[7] Chi phí tài chính**

Số liệu để ghi vào chỉ tiêu này là lấy từ số phát sinh bên nợ tài khoản 911 đối ứng với có tài khoản 635 trong kỳ. Ta lập vùng điều kiện và đặt tên là DKCPTC:

TKGHINO	TKGHICO
911*	635*

[7]= DSUM(SKTM;11;DKCPTC)

✓ **[8] Chi phí bán hàng**

Số liệu để ghi vào chỉ tiêu này ta lấy từ số phát sinh bên nợ tài khoản 911 đối ứng với có tài khoản 641 và có tài khoản 14221 (chi tiết chi phí bán hàng) đối ứng với có tài khoản 911. Ta lập vùng điều kiện và đặt tên là DKCPBH:

TKGHINO	TKGHICO
911*	641*
911*	14221

[8]= DSUM(SKTM;11;DKCPBH)

✓ **[9] Chi phí quản lý doanh nghiệp.**

Số liệu để ghi vào chỉ tiêu này ta lấy từ số phát sinh bên nợ tài khoản 911 đối ứng với có tài khoản 642 và có tài khoản 14222 (chi tiết chi phí quản lý doanh nghiệp) đối ứng với có tài khoản 911. Ta lập vùng điều kiện và đặt tên là DKCPQL:

TKGHINO	TKGHICO
911*	642*
911*	14222

[9]= DSUM(SKTM;11;DKCPQL)

✓ **[10] Lợi nhuận thuần từ hoạt động kinh doanh**

[10]= [5]+[6]-[7]-[8]-[9]

✓ **[11] Thu nhập khác**

Số liệu để ghi vào chỉ tiêu này là lấy từ số phát sinh bên nợ tài khoản 711 đối ứng với có tài khoản 911. Ta lập vùng điều kiện và đặt tên là DKTNK:

TKGHINO	TKGHICO
711*	911*

$$[11]= \text{DSUM}(\text{SKTM};11;\text{DKTNTC})$$

✓ **[12] chi phí khác**

Số liệu để ghi vào chỉ tiêu này là lấy từ số phát sinh bên có tài khoản 811 đối ứng với nợ tài khoản 911. Ta lập vùng điều kiện và đặt tên là DKCPTC:

TKGHINO	TKGHICO
911*	811*

$$[12]= \text{DSUM}(\text{SKTM};11;\text{DKCPTC})$$

✓ **[13] Lợi nhuận khác**

$$[13]= [11] - [12]$$

✓ **[14] Tổng lợi nhuận kế toán trước thuế**

$$[14]= [10] + [13]$$

✓ **[15] Chi phí thuế TNDN hiện hành**

Số liệu để ghi vào chỉ tiêu này được căn cứ vào tổng số phát sinh bên có tài khoản 8211 đối ứng với bên nợ tài khoản 911 trên sổ kế toán chi tiết tài khoản 8211, hoặc căn cứ vào số phát sinh bên nợ tài khoản 8211 đối ứng với có tài khoản 911 trong kỳ báo cáo. Ta lập vùng điều kiện và đặt tên là DKCPTNHH:

TKGHINO	TKGHICO
8211*	911*
911*	8211*

$$[15]= \text{DSUM}(\text{SKTM};11;\text{DKCPTNHH})$$

✓ **[16] Chi phí thuế TNDN hoãn lại.**

Số liệu để ghi vào chỉ tiêu này được căn cứ vào tổng số phát sinh bên có tài khoản 8212 đối ứng với bên nợ tài khoản 911 trên sổ kế toán chi tiết tài khoản 8212, hoặc

căn cứ vào số phát sinh bên nợ tài khoản 8212 đối ứng với có tài khoản 911 trong kỳ báo cáo. Ta lập vùng điều kiện và đặt tên là DKCPTNHL:

TKGHINO	TKGHICO
8212*	911*
911*	8212*

[16]= DSUM(SKTM;11;DKCPTNHL)

✓ [17] Lợi nhuận sau thuế thu nhập doanh nghiệp.

[17]= [14] – ([15] + [16])

BÀI TẬP THỰC HÀNH

Tình hình tài chính Công Ty ABC tháng 03/2010 như sau :

I- Tình hình sản xuất và nguồn vốn của doanh nghiệp vào đầu tháng 03 năm 2010 được thể hiện qua số dư của các tài khoản tổng hợp và chi tiết sau: (ĐVT: VNĐ)

1- Danh mục Tài Khoản kế toán:

MATK	TÊN TÀI KHOẢN	DƯ NỢ ĐẦU KỲ	DƯ CÓ ĐẦU KỲ
1111	Tiền Mặt	30.000.000	
1112	Tiền Gửi Ngân Hàng	170.000.000	
131	Phải thu Khách hàng	50.000.000	
141	Tạm ứng	10.000.000	
1521	Vật liệu chính M1	48.000.000	
1522	Vật liệu phụ N	5.000.000	
1523	Nhiên liệu D	7.000.000	
153	Dụng cụ C1	8.000.000	
154	Giá trị sản phẩm dở dang SPA	2.000.000	
211	Tài sản cố định	670.000.000	
2141	Hao mòn tài sản cố định hữu hình		48.000.000
311	Vay Ngắn hạn Ngân Hàng		172.000.000
331	Phải Trả người cung cấp		70.000.000
411	Nguồn vốn Kinh Doanh		700.000.000
421	Lợi nhuận chưa phân phối		10.000.000
Tổng Cộng:		1.000.000.000	1.000.000.000

2- Sổ công nợ:

MATK	MAKH	TÊN KHÁCH HÀNG	DƯ NỢ ĐẦU KỲ	DƯ CÓ ĐẦU KỲ
131	CTYH1	Công ty H1	50.000.000	
141	NV001	Nguyễn Văn Tư	10.000.000	
331	CTYS1	Công ty S1		70.000.000

3- Vật tư hàng hoá :

MAVT	Tên vật tư	ĐVT	Số lượng	Giá trị (VNĐ)
1521M1	Vật liệu chính M1	Kg	2.000	48.000.000
1522N1	Vật liệu phụ N1	Kg	5.000	5.000.000
1523D1	Nhiên liệu D	Lít	1.400	7.000.000
153C1	Dụng cụ C1	Cái	100	8.000.000

II/ Trong tháng 03/2010 có các nghiệp vụ kinh tế phát sinh sau đây:

1- Ngày 03/03/2010, phiếu xuất kho 100: Xuất 1.000kg vật liệu chính M1 để chế tạo sản phẩm A theo giá thực tế: 24.000.000 đồng.

2- Ngày 03/03/2010, phiếu xuất kho 101: Xuất vật liệu phụ N1 theo giá thực tế :

Trong đó :

- Để chế tạo sản phẩm A : 2.000kg, giá trị : 2.000.000 đồng.

- Để phục vụ sản xuất chung: 500kg, giá trị : 500.000 đồng.
 - Để phục vụ bán hàng: 500kg, giá trị : 500.000 đồng.
- 3- Ngày 03/03/2010, phiếu xuất kho 102: Xuất 100 lít nhiên liệu D1, giá trị: 500.000 đồng sử dụng tại phân xưởng sản xuất.
- 4- Ngày 03/03/2010, phiếu xuất kho số 103: Xuất kho dụng cụ C1 dùng cho phân xưởng sản xuất số lượng là 60 cái, giá trị 4.800.000 đồng và phân bổ dần trong 12 tháng.
- 5- Ngày 15/03/2010, chứng từ ghi sổ 2KH: Trích khấu hao tài sản cố định : 2.400.000 đồng.
 Trong đó :
- Khấu hao máy móc thiết bị,... : 200.000 đồng.
 - Khấu hao nhà, phân xưởng : 200.000 đồng.
 - Khấu hao TSCĐ bộ phận bán hàng : 800.000 đồng.
 - Khấu hao TSCĐ chung toàn doanh nghiệp: 1.200.000 đồng.
- 6- Ngày 20/03/2010, chứng từ ghi sổ số 2TL: Căn cứ bảng chấm công ở các bộ phận, xác định tiền lương phải trả công nhân viên: 20.000.000
 Trong đó :
- Công nhân trực tiếp sản xuất : 10.000.000 đồng.
 - Nhân viên phân xưởng : 4.000.000 đồng.
 - Nhân viên bán hàng : 2.000.000 đồng.
 - Nhân quản lý doanh nghiệp : 4.000.000 đồng.
- 7- Ngày 23/03/2010, chứng từ ghi sổ số 2BYC: Trích bảo hiểm xã hội, bảo hiểm y tế và kinh phí công đoàn theo tỷ lệ quy định trên tiền lương phải trả công nhân viên.
- 8- Ngày 24/03/2010, chứng từ số 2D: Chi phí tiền điện mua ngoài chưa trả tiền người cung cấp (Công ty điện lực) : 550.000, trong đó thuế GTGT là 10%.
 Trong đó :
- Phân xưởng sản xuất : 400.000 đồng.
 - Bộ phận bán hàng : 50.000 đồng.
 - Bộ máy quản lý : 50.000 đồng.
- 9- Ngày 24/03/2010, chứng từ số 2N: Chi phí Tiền nước mua ngoài chưa trả tiền người cung cấp (Công ty cấp thoát nước) : 315.000, trong đó thuế GTGT là 5%.
 Trong đó :
- Phân xưởng sản xuất : 200.000 đồng.
 - Bộ phận bán hàng : 50.000 đồng.
 - Bộ máy quản lý : 50.000 đồng.
- 10- Ngày 24/03/2010, phiếu chi TM số 201: thanh toán tiền điện thoại : 3.080.000 đồng, trong đó thuế GTGT là 280.000 đồng, phân bổ cho các đối tượng sử dụng :
- Phân xưởng sản xuất : 500.000 đồng.
 - Bộ phận bán hàng : 400.000 đồng.
 - Bộ máy quản lý : 1.900.000 đồng.
- 11- Ngày 24/03/2010, Phiếu chi TM số 202: chi tiền mua trái phiếu kho bạc có mệnh giá 10.000.000 đồng, kỳ hạn 5 năm, lãi suất 6%/Năm, Lãnh định kỳ hàng năm.
- 12- Ngày 25/03/2010, chứng từ ghi sổ số 2TLQ: Tạm trích quỹ theo bảng kê sau :
- Quỹ đầu tư phát triển : 5.000.000 đồng.
 - Quỹ khen thưởng: 2.000.000 đồng.
 - Quỹ phúc lợi: 3.000.000 đồng.

- 13- Ngày 24/03/2010, phiếu chi TM số 203: chi tiền mặt 1.000.000 đồng để ủng hộ UBND phường làm công tác xã hội, vệ sinh môi trường,...
- 14- Ngày 25/03/2010, phiếu chi TM số 204: chi tiền tổ chức cho cán bộ công nhân viên đi tham quan di tích lịch sử tại Huế là: 1.800.000 đồng.
- 15- Ngày 27/03/2010, kế toán xác định giá thành sản phẩm A hoàn thành và tiến hành nhập kho – phiếu nhập kho số 200. Biết rằng số lượng sản phẩm hoàn thành là 100 thành phẩm A.
- 16- Ngày 28/03/2010, phiếu xuất kho số 104: Xuất bán cho Công ty H1 50 thành phẩm A, Công ty H1 chấp nhận với giá bán chưa thuế bằng 2 lần giá vốn, Thuế suất thuế GTGT là 10% và đã thanh toán bằng tiền gửi ngân hàng (giấy báo có số 200).
- 17- Cuối tháng, Kế toán xác định kết quả kinh doanh, thuế GTGT và thuế TNDN phải nộp. Biết thuế suất thuế TNDN là 25%.

YÊU CẦU: Sử dụng phần mềm EXCEL để lập các loại sổ sách, báo cáo kế toán phản ánh các nghiệp vụ trên.

☞ Biết rằng: Theo số liệu kiểm kê cuối tháng, xác định có giá trị sản phẩm dở dang sản phẩm A giá trị là: 800.000 đồng. Doanh nghiệp hạch toán vật tư tồn kho theo phương pháp kê khai thường xuyên, hạch toán thuế GTGT theo phương pháp khấu trừ thuế.

TÀI LIỆU THAM KHẢO

[1] Vũ Duy Sanh, ThS. Nguyễn Ngọc Dung, TS. Võ Văn Nhị (2003), “**Ứng dụng EXCEL tự động hoá Công tác Kế toán**”, Nhà Xuất bản Tài chính.

[2] T.S. Bùi Văn Dương, PGS Võ Văn Nhị, Ths Đặng Văn Sáng, KS Nguyễn Ngọc Hiến (2006), “**Hướng dẫn thực hành sổ kế toán lập BCTC & báo cáo thuế trên Excel**”, Nhà xuất bản Lao Động – Xã Hội.